code:
;頭を撫でる

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME321
RESULTS:0 = 頭を撫でる
RESULTS:1 = 頭を撫でさせられる
RESULTS:2 = 頭を撫でてもらう
RESULTS:3 = 頭を撫でられる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE321
;共通部分
CALL COM_ABLE_COMMON(321)
SIF RESULT == 0
	RETURN 0
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM != 1
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM321
;実行できるかの判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	LOCAL:0 = 0
	IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
		LOCAL:0 = 5
	ENDIF
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5 + LOCAL:0, 10 - LOCAL:0)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

;固定で入るソース
SOURCE:(MPLY:0):接触 = 250
SOURCE:(MTAR:0):接触 = 250

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -10
SOURCE:(MTAR:0):歓楽 = -10
CALL ADD_SOURCE_KANRAKU(MPLY:0, 100)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 100)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 130)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 130)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 150, 150)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 150, 150)

IF !TALENT:(MTAR:0):幼児 && !TALENT:(MTAR:0):幼稚 && !TALENT:(MTAR:0):プライド低い
	IF TALENT:(MTAR:0):プライド高い
		SOURCE:(MTAR:0):反感 = 150
		TIMES SOURCE:(MTAR:0):歓楽, 0.50
	ELSE
		SOURCE:(MTAR:0):反感 = 60
		TIMES SOURCE:(MTAR:0):歓楽, 0.80
	ENDIF
ENDIF

IF TALENT:(MPLY:0):母性
	TIMES SOURCE:(MTAR:0):愛情, 1.50
ENDIF

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.20
	TIMES SOURCE:(MPLY:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	TIMES SOURCE:(MTAR:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):反感, 2.00
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TCVAR:(MTAR:0):HOムード値 -= 5
	;成功
ELSEIF TFLAG:コマンド成否 == 0
	TCVAR:(MTAR:0):HOムード値 += 2
	;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):反感, 0.50
	TCVAR:(MTAR:0):HOムード値 += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER321(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 30

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 5)
ENDIF
IF TALENT:(ARG:0):母性
	CALL COM_ORDER_ELEMENT(ARG:0, "母性", 6)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 4)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE321
#DIMS DYNAMIC 実行名
#DIMS DYNAMIC 対象名
#DIMS DYNAMIC 対象_状態
IF IS_EQUIP_PLAYER(MTAR:0, 11, 107)
	対象_状態 '= "対象_口淫"
ELSEIF IS_EQUIP_TARGET(MPLY:0, T_N("膝枕する"))
	対象_状態 '= "対象_膝枕"
ELSEIF IS_EQUIP_PLAYER(MPLY:0, T_N("膝枕する"))
	対象_状態 '= "実行_膝枕"
ENDIF
;キス中
IF IS_EQUIP(MPLY:0, 342)
	LOCALS:0 = キスをしながら
	;地の文カット
	LOCAL:0 = SEARCH_EQUIP_IC(342, MPLY:0, MTAR:0)
	IF LOCAL:0 >= 0
		MEQUIP_CUT:(LOCAL:0) = 1
	ENDIF
ELSE
	LOCALS:0 =
ENDIF

実行名 = %ANAME(MPLY:0)%
対象名 = %ANAME(MTAR:0)%

DEBUGPRINTFORML 対象_状態 = %対象_状態%

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		SELECTCASE 対象_状態
			CASE "対象_口淫"
				PRINTFORM %実行名%は
				PRINTFORM %S_R("一心不乱に/懸命に/ひたすら/",1)%
				PRINTFORM %S_R("ペニス/剛直/亀頭/淫棒",1)%
				PRINTFORM %S_R("に舌を這わせて/を舐め取って/に舌を絡めて/を舐め回して",1)%
				PRINTFORMW 奉仕する%対象名%の頭を撫でた
				PRINTL
			CASE "実行_膝枕"
				{
					CALLF PRINT_RANDC(
						TEXT_CONCATEC(
						@"%実行名%は", 1
						,@"膝の上に頭を預ける%対象名%の頭をそっと撫でた", 1
						), 1
						,TEXT_CONCATEC(
						@"%実行名%は", 1
						,@"膝の上に頭を乗せる%対象名%の髪をそっと撫でた", 1
						), 1
						)
				}
				PRINTL
			CASEELSE
				PRINTFORM %実行名%は%LOCALS:0%%対象名%
				IF CFLAG:(MTAR:0):好感度 >= 800
					PRINTFORMW を抱き寄せて優しく頭を撫でた
				ELSE
					PRINTFORMW の頭を撫でた
				ENDIF
		ENDSELECT
		;主導権が相手側
	ELSE
		PRINTFORM %対象名%は%LOCALS:0%%実行名%
		IF CFLAG:(MTAR:0):好感度 >= 800
			PRINTFORMW に体を預け、頭を撫でるように要求した
		ELSE
			PRINTFORMW に、頭を撫でて欲しいとねだった
		ENDIF
	ENDIF

	IF TFLAG:コマンド成否 == 1
		IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
			PRINTFORML %対象名%は眼を細め、心地良さそうにしている…
		ELSE
			PRINTFORML %対象名%は照れ臭そうに、笑みを浮かべている…
		ENDIF
	ELSEIF TFLAG:コマンド成否 == 0
		IF TALENT:(MTAR:0):幼児 || TALENT:(MTAR:0):幼稚
			PRINTFORML %ANAME(MTAR:0)%は嬉しそうにしている…
		ELSE
			PRINTFORML %ANAME(MTAR:0)%は頬を紅潮させ、恥ずかしそうにしている…
		ENDIF
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML %ANAME(MTAR:0)%は子供扱いされて不機嫌そうだ…
	ENDIF
	;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORM %ANAME(MTAR:0)%は%LOCALS:0%%ANAME(MPLY:0)%に
		IF CFLAG:(MPLY:0):好感度 >= 800
			PRINTFORMW 抱きついて、甘えながら頭を撫でてもらった
		ELSE
			PRINTFORMW 頭を撫でてもらった
		ENDIF

		IF TFLAG:コマンド成否 == 1
			IF TALENT:(MPLY:0):母性
				PRINTFORML %ANAME(MPLY:0)%は我が子を愛おしむかの様に、撫でてくれた…
			ELSE
				PRINTFORML %ANAME(MPLY:0)%は笑顔で優しく撫でてくれた…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == 0
			IF TALENT:(MPLY:0):母性
				PRINTFORML %ANAME(MPLY:0)%は笑顔で優しく、ただ撫でてくれた…
			ELSE
				PRINTFORML %ANAME(MPLY:0)%は呆れた様に苦笑しながらも、優しく撫でてくれた…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML %ANAME(MPLY:0)%は心底呆れたような表情でため息をついた…
		ENDIF
		;主導権が相手側
	ELSE
		PRINTFORM %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%
		IF CFLAG:(MPLY:0):好感度 >= 800
			PRINTFORMW を抱き寄せると、甘やかすように優しく頭を撫でた
		ELSE
			PRINTFORMW の頭を撫でてきた
		ENDIF

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
		ELSEIF TFLAG:コマンド成否 == 0
			PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応を見て楽しんでいる…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML が、%ANAME(MPLY:0)%はすぐに飽きてしまったようだ…
		ENDIF
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST321
#DIMS DYNAMIC 対象_状態
FOR LOCAL, 0, MTAR_NUM
	IF IS_SERVEAT(MTAR:LOCAL, "口", "フェラ") && CHECK_EMOTION(MTAR:LOCAL, "GOOD")
		CALLF TAG_PRINT(@"@C:{カラー_ピンク}@%ONOMATOPE_PRESET("フェラ音長", (RAND:2), 0)%@/C@", 1)
		PRINTL
		;あまりピンクを多用すると読みにくい気がする。
		CALLF TAG_PRINT(@"@C:{カラー_白}@%ANAME(MTAR:LOCAL)%は%ANAME(MPLY:0)%%S_R("に奉仕を褒められ/に頭を撫でられながら/の手に頭を預けて")%、%S_R("嬉しそうに/頬を紅潮させ/うっとりした様子で/表情を弛緩させ/熱い吐息をもらしながら")%%S_R("一層夢中に/より丹念に/さらに積極的に")%%語彙_ペニス()%%S_R("を舐め回して/に舌を這わせて/にむしゃぶりついて/に奉仕して")%いる@C@", 1)
	ENDIF
NEXT
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------
;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN321
IF FLAG:調教モード != 調教_居住区
	RETURN コマンド_日常
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;髪を梳く

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME322
RESULTS:0 = 髪を梳く
RESULTS:1 = 髪梳きさせられる
RESULTS:2 = 髪を梳いてもらう
RESULTS:3 = 髪梳きされる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE322
;共通部分
CALL COM_ABLE_COMMON(322)
SIF RESULT == 0
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM322
;実行できるかの判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
ENDIF

;固定で入るソース
SOURCE:(MPLY:0):接触 = 250
SOURCE:(MTAR:0):接触 = 250

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -10
SOURCE:(MTAR:0):歓楽 = -10
CALL ADD_SOURCE_KANRAKU(MPLY:0, 120)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 120)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 120)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 120)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 70, 70)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 70, 70)

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 1

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.20
	TIMES SOURCE:(MPLY:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	TIMES SOURCE:(MTAR:0):愛情, 0.20
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TFLAG:ムード -= 5
;成功
ELSEIF TFLAG:コマンド成否 == 0

;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER322(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 30

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):母性
	CALL COM_ORDER_ELEMENT(ARG:0, "母性", 3)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 3)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE322
;キス中
IF IS_EQUIP(MPLY:0, 342)
	LOCALS:0 = キスをしながら
	;地の文カット
	LOCAL:0 = SEARCH_EQUIP_IC(342, MPLY:0, MTAR:0)
	IF LOCAL:0 >= 0
		MEQUIP_CUT:(LOCAL:0) = 1
	ENDIF
ELSE
	LOCALS:0 =
ENDIF

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の髪に手櫛を入れた
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%に頼まれて、手櫛で髪を梳かした
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MTAR:0)%は目を閉じて気持ちよさそうにしている…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%にされるまま身を任せている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML 途中で%ANAME(MTAR:0)%の髪を引っ掛けてしまった…
		PRINTFORM %ANAME(MTAR:0)%は目尻に涙を浮かべて
		IF TALENT:(MTAR:0):恥じらい || TALENT:(MTAR:0):臆病 || TALENT:(MTAR:0):大人しい
			PRINTFORML 痛そうに頭を抑えた…
		ELSE
			PRINTFORML %ANAME(MPLY:0)%を睨んでいる…
		ENDIF
	ENDIF
;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に頼んで、%LOCALS:0%手櫛で髪を梳いてもらった
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の髪に手櫛を入れた
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MPLY:0)%は丁寧に%ANAME(MTAR:0)%の髪に指を入れている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML が、%ANAME(MPLY:0)%はすぐに飽きてしまったようだ…
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST322
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN322
RETURN コマンド_日常

repo_name: napo/erafl

license: unknown


code:
;櫛で梳かす

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME323
RESULTS:0 = 櫛で髪を梳かす
RESULTS:1 = 櫛で髪を整えさせられる
RESULTS:2 = 櫛で髪を梳かしてもらう
RESULTS:3 = 櫛で髪を梳かされる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE323
;共通部分
CALL COM_ABLE_COMMON(323)
SIF RESULT == 0
	RETURN 0
SIF !ITEM:櫛
	RETURN 0
;主人公以外が実行する場合、好感度が500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 500
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM323
;実行できるかの判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

;固定で入るソース
SOURCE:(MPLY:0):接触 = 50
SOURCE:(MTAR:0):接触 = 50

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -10
SOURCE:(MTAR:0):歓楽 = -10
CALL ADD_SOURCE_KANRAKU(MPLY:0, 140)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 140)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 120)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 120)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 70, 70)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 70, 70)

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.20
	TIMES SOURCE:(MPLY:0):愛情, 0.20
	TIMES SOURCE:(MTAR:0):歓楽, 0.20
	TIMES SOURCE:(MTAR:0):愛情, 0.20
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TFLAG:ムード -= 5
;成功
ELSEIF TFLAG:コマンド成否 == 0

;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER323(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 25

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):母性
	CALL COM_ORDER_ELEMENT(ARG:0, "母性", 3)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 3)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE323
;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の髪を櫛で梳かした
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に頼まれて、櫛で髪を梳かした
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MTAR:0)%は目を閉じて気持ちよさそうにしている…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%にされるまま身を任せている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML 途中で%ANAME(MTAR:0)%の髪を引っ掛けてしまった…
		PRINTFORM %ANAME(MTAR:0)%は目尻に涙を浮かべて
		IF TALENT:(MTAR:0):恥じらい || TALENT:(MTAR:0):臆病 || TALENT:(MTAR:0):大人しい
			PRINTFORML 痛そうに頭を抑えた…
		ELSE
			PRINTFORML %ANAME(MPLY:0)%を睨んでいる…
		ENDIF
	ENDIF
;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に頼んで、櫛で髪を梳いてもらった
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の髪を櫛で梳かした
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MPLY:0)%はとても楽しそうだ…
	ELSEIF TFLAG:コマンド成否 == 0
		PRINTFORML %ANAME(MPLY:0)%は丁寧に%ANAME(MTAR:0)%の髪を整えている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML が、%ANAME(MPLY:0)%はすぐに飽きてしまったようだ…
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST323
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN323
RETURN コマンド_日常


code:
;甘い言葉

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME324
RESULTS:0 = 甘い言葉をかける
RESULTS:1 = 甘い言葉をねだられる
RESULTS:2 = 甘い言葉をねだる
RESULTS:3 = 甘い言葉を囁かれる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE324
;共通部分
CALL COM_ABLE_COMMON(324)
SIF RESULT == 0
	RETURN 0
;主人公以外が実行する場合、好感度が1500以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 1500
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM324
;実行判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
ENDIF

;好感度500未満なら確実に失敗
IF MPLY:0 == MASTER && CFLAG:(MTAR:0):好感度 < 500
	TFLAG:コマンド成否 = -1
ENDIF

;主導権に応じた優越・受動のソース追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 70, 70)
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 70, 70)

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -50
SOURCE:(MTAR:0):歓楽 = -50
CALL ADD_SOURCE_KANRAKU(MPLY:0, 90)
CALL ADD_SOURCE_KANRAKU(MTAR:0, 90)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 150)
CALL ADD_SOURCE_AIZYOU(MTAR:0, 150)

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.10
	TIMES SOURCE:(MPLY:0):優越, 0.50
	TIMES SOURCE:(MPLY:0):愛情, 0.10
	TIMES SOURCE:(MTAR:0):歓楽, 0.10
	TIMES SOURCE:(MTAR:0):優越, 0.50
	TIMES SOURCE:(MTAR:0):愛情, 0.10
	SOURCE:(MPLY:0):反感 += 100
	SOURCE:(MTAR:0):反感 += 100
	SOURCE:(MPLY:0):不満 += 800
	SOURCE:(MTAR:0):不満 += 800
	TFLAG:ムード -= 8
	;成功
ELSEIF TFLAG:コマンド成否 == 0

	;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):優越, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):優越, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 8
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER324(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 40

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 2)

;好感度
LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)

IF TALENT:(ARG:0):恥薄い
	CALL COM_ORDER_ELEMENT(ARG:0, "恥薄い", 4)
ENDIF
IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -6)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 2)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 7)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE324
;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に歯の浮くような甘い言葉を掛けた

		IF TFLAG:コマンド成否 == 1
			IF TALENT:(MTAR:0):反抗的 || TALENT:(MTAR:0):生意気 || TALENT:(MTAR:0):保守的
				PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の言葉を冷たい態度で受け流したが、嬉しさを隠せず時折笑みをこぼしている…
			ELSE
				IF GROUPMATCH(TALENT:(MTAR:0):性別, 0, 3)
					PRINTFORML %ANAME(MTAR:0)%は顔を真っ赤にしながらも、幸せそうに%ANAME(MPLY:0)%を抱きしめた…
				ELSE
					PRINTFORML %ANAME(MTAR:0)%は顔を真っ赤にしながらも、幸せそうに%ANAME(MPLY:0)%に抱きついた…
				ENDIF
			ENDIF
		ELSEIF TFLAG:コマンド成否 == 0
			IF TALENT:(MTAR:0):反抗的 || TALENT:(MTAR:0):生意気 || TALENT:(MTAR:0):保守的
				PRINTFORML %ANAME(MTAR:0)%は突然何を言っているのかと%ANAME(MPLY:0)%を罵ったが、その表情は満更でもなさそうだ…
			ELSE
				PRINTFORML %ANAME(MTAR:0)%は突然の言葉に、顔を真っ赤にして慌てている…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == -1
			IF TALENT:(MTAR:0):反抗的 || TALENT:(MTAR:0):生意気 || TALENT:(MTAR:0):保守的
				PRINTFORML %ANAME(MTAR:0)%は汚物を見るかのような目で%ANAME(MPLY:0)%を見た…
			ELSE
				PRINTFORML だが、%ANAME(MTAR:0)%に盛大に引かれてしまった…
			ENDIF
		ENDIF
		;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の求めに応じて、思いつく限りの甘い言葉を掛けた

		IF TFLAG:コマンド成否 == 1
			IF TALENT:(MTAR:0):反抗的 || TALENT:(MTAR:0):生意気 || TALENT:(MTAR:0):保守的
				PRINTFORML %ANAME(MTAR:0)%は冷静を装っているが、すっかり顔を真っ赤にして、胸の高鳴りを隠せずにいる…
			ELSE
				PRINTFORML %ANAME(MTAR:0)%は予想以上にドキドキしてしまったらしく、顔を真っ赤にして%ANAME(MPLY:0)%を見つめた…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == 0
			IF TALENT:(MTAR:0):反抗的 || TALENT:(MTAR:0):生意気 || TALENT:(MTAR:0):保守的
				PRINTFORML %ANAME(MTAR:0)%に一応は満足してもらえたようだ…
			ELSE
				PRINTFORML どうやら%ANAME(MTAR:0)%は満足したらしく、%ANAME(MPLY:0)%に向けて微笑んだ…
			ENDIF
		ELSEIF TFLAG:コマンド成否 == -1
			IF TALENT:(MTAR:0):反抗的 || TALENT:(MTAR:0):生意気 || TALENT:(MTAR:0):保守的
				PRINTFORML %ANAME(MTAR:0)%はもっとマトモなことは言えないのかと%ANAME(MPLY:0)%を罵った…
			ELSE
				PRINTFORML だが、%ANAME(MTAR:0)%は反応に困って目を逸らしてしまった…
			ENDIF
		ENDIF
	ENDIF

	;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の求めに応じて、思いつく限りの甘いセリフを掛けてきた

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MTAR:0)%が%ANAME(MPLY:0)%を褒めると、%ANAME(MTAR:0)%は幸せそうに微笑んだ…
		ELSEIF TFLAG:コマンド成否 == 0
			PRINTFORML %ANAME(MPLY:0)%は恥ずかしさのあまり顔を真っ赤にしている…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML だが、%ANAME(MPLY:0)%は恥ずかしさのあまり黙りこんでしまった…
		ENDIF
		;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に歯の浮くような甘い言葉を掛けてきた

		IF TFLAG:コマンド成否 == 1
			PRINTFORML %ANAME(MTAR:0)%と%ANAME(MPLY:0)%は雰囲気に流されるまま、お互いをじっと見つめ合った…
		ELSEIF TFLAG:コマンド成否 == 0
			PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応を見て楽しんでいる…
		ELSEIF TFLAG:コマンド成否 == -1
			PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応が薄いことに拗ねてしまった…
		ENDIF
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST324
;日常コマンドに共通の地の文
;CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN324
RETURN コマンド_日常


code:
;見つめ合う

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME325
VARS COM_NAME

IF MTAR == -1 || MPLY == MTAR
	COM_NAME '= "眺める"
	RESULTS:0 = %COM_NAME%
	RESULTS:1 = %COM_NAME%
	RESULTS:2 = %COM_NAME%
	RESULTS:3 = %COM_NAME%
	RESULTS:4 = %COM_NAME%
	RESULTS:5 = %COM_NAME%
ELSE
	SELECTCASE COM_OPTION:325
		CASE 0
			COM_NAME '= "顔を見つめる"
		CASE 1
			COM_NAME '= "胸を見つめる"
		CASE 2
			COM_NAME '= "股を見つめる"
		CASE 3
			COM_NAME '= "尻を見つめる"
	ENDSELECT
	RESULTS:0 = %COM_NAME%
	RESULTS:1 = %COM_NAME%
	RESULTS:2 = %COM_NAME%
	RESULTS:3 = %COM_NAME%
	RESULTS:4 = %COM_NAME%
	RESULTS:5 = %COM_NAME%
ENDIF

;-------------------------------------------------
;オプションを持つコマンドの場合オプション数を指定
;-------------------------------------------------
@COM_HAS_OPTION325
IF (FLAG:調教モード == 調教_居住区 || (FLAG:クエストモード && FLAG:クエスト_交流コマンド許可)) && MPLY == MASTER
	IF MTAR_NUM
		RETURN 3
	ELSE
		RETURN 0
	ENDIF
ELSE
	RETURN 0
ENDIF

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE325
;共通部分
CALL COM_ABLE_COMMON(325)
SIF RESULT == 0
	RETURN 0
;主人公がプレイヤーのとき限定
IF GROUPMATCH(FLAG:調教モード, 調教_居住区, 調教_クエスト)
	SIF MPLY:0 != MASTER
		RETURN 0
ENDIF
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM > 1
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM325
#DIMS DYNAMIC 視線
#DIMS DYNAMIC LIST_CHARA
#DIM  DYNAMIC CHARA_NUM
#DIM  DYNAMIC CHARA_ID
#DIMS DYNAMIC MODULE_NAME
#DIMS DYNAMIC TEXT
#DIMS DYNAMIC INPUT_VALUES
#DIM  DYNAMIC L_CNT
#DIMS DYNAMIC STRINGS

;主人公が実行する場合
IF (FLAG:調教モード == 調教_居住区 || (FLAG:クエストモード && FLAG:クエスト_交流コマンド許可)) && MPLY == MASTER
	;眺めるコマンドの場合
	IF !MTAR_NUM
		COM_OPTION:325 = 0
		CALL HO_GET_MEMBER_IN_ROOM(HO_MASTER現在地, 0)
		LIST_CHARA = %RESULTS%
		CHARA_NUM = RESULT
		IF !CHARA_NUM
			PRINTFORML 歌や踊りを披露している仲間がいれば鑑賞できる
			PRINTFORMW 今は対象がいないようだ
			RETURN 0
		ENDIF
		;ボタン生成
		FOR LOCAL, 0, CHARA_NUM
			CHARA_ID = LIST_GETINT(LIST_CHARA, LOCAL)
			SIF !C_COND("設備属性:鑑賞",,CHARA_ID)
				CONTINUE
			;既に同じ設備にいる
			SIF C_COND_MAP_GET(CHARA_ID, "設備IDX") == HO_MASTER_USE_MODULE_LIST_IDX
				CONTINUE
			CALL HO_IS_TARGET_VISIBLE(MPLY, CHARA_ID)
			SIF !RESULT
				CONTINUE
			MODULE_NAME '= C_COND_MAP_GETS(CHARA_ID, "設備名")
			LOCALS '= @"%MODULE_NAME%にいる%ANAME(CHARA_ID)%"
			TEXT += @"<button value='[CHARA_ID:{CHARA_ID}]'>[%LOCALS%]</button>" + "
"
			L_CNT ++
		NEXT
		IF !L_CNT
			PRINTFORML 歌や踊りを披露している仲間がいれば鑑賞できる
			PRINTFORMW 今は対象がいないようだ
			RETURN 0
		ENDIF
		TEXT += @"<button value='[CANCEL:1]'>[やっぱりやめる]</button>" + "
"
		HTML_PRINT TEXT
	ENDIF
ELSE
	;主人公以外実行する場合はランダム
	COM_OPTION:325 = RAND(1, 3)
ENDIF

IF !MTAR_NUM
	BINPUTS
	CLEARLINE 1
	PRINTL
	INPUT_VALUES = %RESULTS%
	SIF DIC_CONTAINSKEY(INPUT_VALUES, "CANCEL")
		RETURN 0
	CHARA_ID = TOINT(DIC_GET(INPUT_VALUES, "CHARA_ID"))
	CALL ADD_MTAR(CHARA_ID)
	CFLAG:CHARA_ID:調教参加フラグ = 1
	TCVAR:CHARA_ID:調教一時参加扱い = 1
	TFLAG:参加者数 ++
	CALL C_COND_MAP_SET(MPLY:0, "視線", "鑑賞")
ELSE
	SELECTCASE COM_OPTION:325
		CASE 0
			CALL C_COND_MAP_SET(MPLY:0, "視線", "顔")
		CASE 1
			CALL C_COND_MAP_SET(MPLY:0, "視線", "胸")
		CASE 2
			CALL C_COND_MAP_SET(MPLY:0, "視線", "股")
		CASE 3
			CALL C_COND_MAP_SET(MPLY:0, "視線", "尻")
		CASE 4
			RETURN 0
	ENDSELECT
ENDIF

;実行判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

;コマンドの成否をTFLAG:コマンド成否にセット
CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)

;好感度500未満なら確実に失敗
IF MPLY:0 == MASTER && CFLAG:(MTAR:0):好感度 < 500
	TFLAG:コマンド成否 = -1
ENDIF

視線  = %C_COND_MAP_GETS(MPLY:0, "視線")%

;プレイヤーの処理
;-------------------------------------------------
;主導権に応じた優越・受動のソース追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 50, 50)

;親密に応じた歓楽のソース追加
SOURCE:(MPLY:0):歓楽 = -50
CALL ADD_SOURCE_KANRAKU(MPLY:0, 70)
SOURCE:(MPLY:0):歓楽 = MAX(SOURCE:(MPLY:0):歓楽, 0)
IF 視線 == "顔"
	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MPLY:0, 165)
ELSE
	IF IS_SEXUAL(MPLY:0, MTAR:0)
		SOURCE:(LOCAL:2):性行動 = 90
	ENDIF
ENDIF

;ターゲットの処理
;-------------------------------------------------
;主導権に応じた優越・受動のソース追加
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 50, 50)

IF 視線 == "顔"
	;親密に応じた歓楽のソース追加
	SOURCE:(MTAR:0):歓楽 = -50
	CALL ADD_SOURCE_KANRAKU(MTAR:0, 70)
	SOURCE:(MTAR:0):歓楽 = MAX(SOURCE:(MTAR:0):歓楽, 0)
	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MTAR:0, 165)
ELSE
	IF (視線 == "胸" && C_COND("露出部位:胸",,MTAR:0)) || (視線 == "股" && C_COND("露出部位:秘所",,MTAR:0)) || C_COND("露出部位:全裸",,MTAR:0)
		SOURCE:(MTAR:0):露出 = 150
		SOURCE:(MTAR:0):逸脱 = 50
		SOURCE:(MTAR:0):感応 = 100 + SOURCE:(MTAR:0):感応
		SELECTCASE 視線
			CASE "胸"
				SOURCE:(MTAR:0):快Ｂ = SENSE_HOUSHI(MTAR:0, MTAR:0, 5)
			CASE "股"
				SOURCE:(MTAR:0):快Ｃ = SENSE_HOUSHI(MTAR:0, MTAR:0, 5)
		ENDSELECT
	ELSE
		SOURCE:(MTAR:0):露出 = 50
		SOURCE:(MTAR:0):逸脱 = 25
		SOURCE:(MTAR:0):感応 = 100 + SOURCE:(MTAR:0):感応 / 2
	ENDIF
ENDIF

;プレイヤー・ターゲット共通処理
;-------------------------------------------------
;失敗
IF TFLAG:コマンド成否 == -1
	IF 視線 == "顔"
		TIMES SOURCE:(MPLY:0):歓楽, 0.10
		TIMES SOURCE:(MPLY:0):優越, 0.50
		TIMES SOURCE:(MPLY:0):愛情, 0.10
		TIMES SOURCE:(MTAR:0):歓楽, 0.10
		TIMES SOURCE:(MTAR:0):優越, 0.50
		TIMES SOURCE:(MTAR:0):愛情, 0.10
		SOURCE:(MPLY:0):反感 += 60
		SOURCE:(MTAR:0):反感 += 60
		SOURCE:(MPLY:0):不満 += 500
		SOURCE:(MTAR:0):不満 += 500
		TCVAR:(MTAR:0):HOムード値 -= 5
	ELSE
		TIMES SOURCE:(MPLY:0):歓楽, 0
		TIMES SOURCE:(MPLY:0):優越, 0.30
		TIMES SOURCE:(MPLY:0):愛情, 0
		TIMES SOURCE:(MTAR:0):歓楽, 0
		TIMES SOURCE:(MTAR:0):優越, 0.50
		SOURCE:(MTAR:0):反感 += 150
		SOURCE:(MPLY:0):不満 += 900
		SOURCE:(MTAR:0):不満 += 900
		TCVAR:(MTAR:0):HOムード値 -= 10
	ENDIF
	;成功
ELSEIF TFLAG:コマンド成否 == 0

	;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):優越, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):優越, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TCVAR:(MTAR:0):HOムード値 += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 0

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE325
#DIMS DYNAMIC 視線
#DIMS DYNAMIC 衣装描写
#DIM  DYNAMIC 態度値
#DIM  DYNAMIC 向性値
#DIMS DYNAMIC 行動

視線  = %C_COND_MAP_GETS(MPLY:0, "視線")%
SELECTCASE 視線
	CASE "顔"
		SELECTCASE TFLAG:コマンド成否
			CASE 1
				PRINTFORMW %ANAME(MPLY:0)%と%ANAME(MTAR:0)%は目を合わせてじっと見つめ合った
				PRINTFORML 二人の間に甘い空気が流れた…
			CASE 0
				PRINTFORMW %ANAME(MPLY:0)%と%ANAME(MTAR:0)%は目を合わせてじっと見つめ合った
				IF C_COND("素質:反抗的, 素質:生意気, 素質:保守的","OR", MTAR:0)
					PRINTFORML しばらくすると%ANAME(MTAR:0)%は照れるようにして目を逸らした…
				ELSE
					PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に優しく微笑んでいる…
				ENDIF
			CASE -1
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%と目を合わせて見つめ合おうとした
				IF C_COND("素質:反抗的, 素質:生意気, 素質:保守的","OR", MTAR:0)
					PRINTFORML ……が、%ANAME(MTAR:0)%に気色悪そうな目で見られてしまった
				ELSE
					PRINTFORML ……が、%ANAME(MTAR:0)%はよくわからずに困惑した表情を浮かべた
				ENDIF
		ENDSELECT
	CASE "胸"
		PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の胸元をまじまじと見つめた
		IF C_COND("露出_胸:下着",, MTAR:0)
			衣装描写 = %C_COND_MAP_GETS(MTAR:0, "被覆衣装描写_胸")%
			IF C_COND("快感:快Ｂ:>=9000",, MTAR:0)
				CALLF TAG_PRINT(@"%衣装描写%の下から@C:{カラー_ピンク}@%STR_BODY("乳首：欲情：感度", MTAR:0)%@/C@が%S_R("はっきりと主張している/くっきりと浮き出ている")%のが見て取れる", 1)
			ELSE
				PRINTFORML %衣装描写%に覆われた%STR_BODY("胸", MTAR:0)%が見て取れる
			ENDIF
		ELSEIF C_COND("露出_胸:露出",, MTAR:0)
			CALLF TAG_PRINT(@"@C:{カラー_ピンク}@%STR_BODY("乳首：欲情：感度", MTAR:0)%@/C@が%S_R(@"顕になっている/%ANAME(MPLY:0)%の目に晒されている")%…", 1)
		ENDIF
		PRINTL
		IF IS_FALL(MTAR:0) && C_COND("快感ＬＶ:欲情:>7",, MTAR:0)
			IF C_COND("素質:プライド高い, 素質:小悪魔",, MTAR:0)
				LOCALS '= @"いたずらな笑みを浮かべると、%ANAME(MPLY:0)%の視線に答えるように"
			ELSE
				LOCALS '= @"気恥ずかしげに頬を紅潮させつつも、笑みを浮かべて%ANAME(MPLY:0)%に"
			ENDIF
			PRINTFORML %ANAME(MTAR:0)%は%LOCALS%胸を強調して見せた
		ELSEIF TFLAG:コマンド成否 ==-1
			IF IS_FALL(MTAR:0)
				LOCALS '= "少し嫌な顔をして、"
			ELSE
				LOCALS '= "はっきりと嫌悪の表情を浮かべて、"
			ENDIF
			PRINTFORML %ANAME(MTAR:0)%は%LOCALS%%ANAME(MPLY:0)%の視線から逃れるように身をよじった
		ELSE
			PRINTFORML %ANAME(MTAR:0)%は困った表情を浮かべてうつむくと、%ANAME(MPLY:0)%の視線から逃れるように身をよじった
		ENDIF
	CASE "股"
		PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の下腹部をまじまじと見つめた
		IF C_COND("露出_秘部:下着",, MTAR:0)
			衣装描写 = %C_COND_MAP_GETS(MTAR:0, "被覆衣装描写_秘部")%
			IF C_COND("快感ＬＶ:潤滑:>=8, 快感ＬＶ:欲情:>7","AND", MTAR:0) && IS_FEMALE(MTAR:0)
				CALLF TAG_PRINT(@"愛液%S_R("でぐしょぐしょに濡れた/でびしょびしょになった/ですっかり濡れそぼってしまった")%%衣装描写%の下で、@C:{カラー_ピンク}@%STR_BODY("クリ：欲情：感度", MTAR:0)%@/C@を疼かせているのが伺える…", 1)
			ELSEIF C_COND("快感ＬＶ:潤滑:>=8, 快感ＬＶ:欲情:>7","AND", MTAR:0) && IS_MALE(MTAR:0)
			ELSE
				PRINTFORML %衣装描写%が顕になっている
			ENDIF
		ELSEIF C_COND("露出_秘部:露出",, MTAR:0) && IS_FEMALE(MTAR:0)
			IF C_COND("快感ＬＶ:潤滑:>=8, 快感ＬＶ:欲情:>7","AND", MTAR:0) && IS_FEMALE(MTAR:0)
				CALLF TAG_PRINT(@"%S_R("ぬめりを帯びた愛液を滴らせる/とろとろと愛蜜を滴らす/糸を引かせて愛液を垂らす")%@C:{カラー_ピンク}@%STR_BODY("クリ：欲情：感度", MTAR:0)%@/C@が顕になり、羞恥に疼いている…", 1)
			ELSE
				CALLF TAG_PRINT(@"@C:{カラー_ピンク}@%STR_BODY("クリ：欲情：感度", MTAR:0)%@/C@が%S_R(@"顕になっている/%ANAME(MPLY:0)%の目に晒されている")%…", 1)
			ENDIF
		ENDIF
		PRINTL
		IF IS_FALL(MTAR:0) && C_COND("快感ＬＶ:欲情:>7",, MTAR:0)
			IF C_COND("素質:プライド高い, 素質:小悪魔",, MTAR:0)
				LOCALS '= @"いたずらな笑みを浮かべると、%ANAME(MPLY:0)%の視線に答えるように"
			ELSE
				LOCALS '= @"気恥ずかしげに頬を紅潮させつつも、笑みを浮かべて%ANAME(MPLY:0)%の前で"
			ENDIF
			IF C_COND("衣装属性:秘部:スカート",,MTAR)
				{
					CALLF PRINT_RANDC(
						@"%ANAME(MTAR)%は人目を盗むようにこっそりと、そっとスカートをつまみ上げて脚のラインを強調してみせた", !C_COND("設備属性:遮蔽",,MTAR)
						,@"%ANAME(MTAR)%はそっとスカートをつまみ上げて、太もものラインを強調してみせた", C_COND("設備属性:遮蔽",,MTAR) && ACCEPT_EACH()
						)
				}
				PRINTL
			ELSE
				PRINTFORML %ANAME(MTAR:0)%は%LOCALS%脚を広げて挑発してみせた
			ENDIF
		ELSEIF TFLAG:コマンド成否 ==-1
			IF IS_FALL(MTAR:0)
				LOCALS '= "少し嫌な顔をして、"
			ELSE
				LOCALS '= "はっきりと嫌悪の表情を浮かべて、"
			ENDIF
			PRINTFORML %ANAME(MTAR:0)%は%LOCALS%%ANAME(MPLY:0)%の視線から逃れるように身をよじった
		ELSE
			PRINTFORML %ANAME(MTAR:0)%は困った表情を浮かべてうつむくと、%ANAME(MPLY:0)%の視線から逃れるように身をよじった
		ENDIF
	CASE "尻"
		PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%STR_BODY("尻", MTAR:0)%をまじまじと見つめた
		IF IS_FALL(MTAR:0) && C_COND("快感ＬＶ:欲情:>7",, MTAR:0)
			IF C_COND("素質:プライド高い, 素質:小悪魔",, MTAR:0)
				LOCALS '= @"いたずらな笑みを浮かべると、%ANAME(MPLY:0)%の視線に答えるように"
			ELSE
				LOCALS '= @"気恥ずかしげに頬を紅潮させつつも、笑みを浮かべて%ANAME(MPLY:0)%の前で"
			ENDIF
			PRINTFORML %ANAME(MTAR:0)%は%LOCALS%尻を強調してみせた
		ELSEIF TFLAG:コマンド成否 ==-1
			IF IS_FALL(MTAR:0)
				LOCALS '= "少し嫌な顔をして、"
			ELSE
				LOCALS '= "はっきりと嫌悪の表情を浮かべて、"
			ENDIF
			PRINTFORML %ANAME(MTAR:0)%は%LOCALS%%ANAME(MPLY:0)%の視線から逃れるように身をよじった
		ELSE
			PRINTFORML %ANAME(MTAR:0)%は困った表情を浮かべてうつむくと、%ANAME(MPLY:0)%の視線から逃れるように身をよじった
		ENDIF
	CASE "鑑賞"
		CALLF ATTITUDE(MTAR, "娯楽")
		態度値 = RESULT
		向性値 = RESULT:1
		行動 = %C_COND_MAP_GETS(MTAR, "行動")%
		LOCALS '= ""
		LOCALS += @"%ANAME(MPLY)%は"
		LOCALS += @"%C_COND_MAP_GETS(MTAR, "設備名")%で"
		IF 行動 == "演奏中"
			LOCALS += @"曲を披露している"
		ELSEIF 行動 == "演舞中"
			LOCALS += @"踊りを披露している"
		ENDIF
		LOCALS += @"%ANAME(MTAR)%を"
		LOCALS += @"遠巻きに見つめた"
		PRINTFORML %LOCALS%
		;視線に対する反応
		LOCALS '= ""
		LOCALS += @"%ANAME(MTAR)%は視線に気がつくと、"
		IF !IS_ESTRUS(6)
			{
				LOCALS += TEXT_RANDC(
					"にこやかに微笑んでパフォーマンスしてみせた", 態度値 > 2 && 向性値 > 0
					,"ウィンクして応じた", 態度値 >= 2 && 向性値 >= 0
					,"気恥ずかしげに微笑んでみせた", 態度値 >= 0 && 向性値 <= 0
					,"愉快そうに手を振ってみせた", 態度値 >= 0 && 向性値 >= 2
					,"特に反応も示さず再び芸に没頭しはじめた", 態度値 <= 1
					,"少し顔をしかめてこちらを一瞥し、再び芸に没頭しはじめた", 態度値 <= -2 && 向性値 >= 0
					,"さっと目をそらして再び芸に没頭しはじめた", 態度値 <= -2 && 向性値 <= 0
					)
			}
		ELSE
			{
				LOCALS += TEXT_RANDC(
					"上気した頬に笑みを浮かべてパフォーマンスしてみせた", 態度値 > 2 && 向性値 > 0
					,"艶めかしい笑みを浮かべて応じた", 態度値 >= 2 && 向性値 >= 0
					,"顔を赤らめてうつむいた", 態度値 >= 0 && 向性値 <= 0
					,"情欲に満ちた瞳を輝かせて笑みを浮かべた", 態度値 >= 0 && 向性値 >= 2
					,"顔を赤らめてうつむいた", 態度値 <= 1
					,"上気した顔をしかめてこちらを一瞥し、再び芸に没頭しはじめた", 態度値 <= -2 && 向性値 >= 0
					,"火照りを悟られぬようさっと目をそらし、再び芸に没頭しはじめた", 態度値 <= -2 && 向性値 <= 0
					)
			}
		ENDIF
		PRINTFORML %LOCALS%
		;ローター、バイブの描写
		IF IS_EQUIP_TARGET(MTAR, T_N("バイブ"), T_N("アナルバイブ"), T_N("ローター"))
			PRINTL
			LOCALS '= ""
			{
				LOCALS += TEXT_RANDC(
					"歌を披露しながらも、", 行動 == "演奏中"
					,@"%S_R("舞を披露する最中にも/ステップを踏む最中にも")%、", 行動 == "演舞中"
					)
			}
			{
				LOCALS += TEXT_RANDC(
					@"%ANAME(MTAR)%はローターの与える快楽に時折身を震わせている", IS_EQUIP_TARGET(MTAR, T_N("ローター"))
					,@"%ANAME(MTAR)%はバイブの与える快楽にひくひくと身を震わせている", IS_EQUIP_TARGET(MTAR, T_N("バイブ"), T_N("アナルバイブ"))
					,@"%ANAME(MTAR)%はローターの与える快楽を堪えきれておらず、音色にかすかに切なげな嬌声が混じっている",IS_EQUIP_TARGET(MTAR, T_N("ローター")) && 行動 == "演奏中" && C_COND("快感ＬＶ:Ｃ感:>4",,MTAR)
					,@"%ANAME(MTAR)%はバイブの与える快楽を堪えきれておらず、音色にあきらかに切なげな嬌声が混じっている",IS_EQUIP_TARGET(MTAR, T_N("バイブ"), T_N("アナルバイブ")) && 行動 == "演奏中" && C_COND("快感ＬＶ:Ｃ感:>4,快感ＬＶ:Ａ感:>4","OR",MTAR)
					)
			}
			SETCOLOR カラー_ピンク
			PRINTFORML %LOCALS%
			PRINTL
			RESETCOLOR
		ENDIF
ENDSELECT

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST325
#DIMS 視線
視線  = %C_COND_MAP_GETS(MPLY:0, "視線")%
IF IS_EQUIP_PLAYER(MPLY:0, T_N("Ｐ自慰")) && GROUPMATCH(視線, "胸", "尻")
	CALLF TAG_PRINT(@"@C:{カラー_ピンク}@%ANAME(MPLY:0)%は%ANAME(MTAR:0)%の%STR_BODY(視線, MTAR:0)%に一層劣情をたぎらせ、%語彙_ペニス()%をしごく手を早めた…@/C@", 1)
ENDIF
;日常コマンドに共通の地の文
;CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN325
IF FLAG:調教モード != 調教_居住区
	IF TCVAR:(MPLY:0):被奉仕者フラグ == 奉仕_見抜き
		RETURN コマンド_ウフフ
	ELSE
		RETURN コマンド_日常
	ENDIF
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;お腹の音を聴く

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME326
RESULTS:0 = お腹の音を聴く
RESULTS:1 = お腹の音を聴かされる
RESULTS:2 = お腹の音を聴かせる
RESULTS:3 = お腹の音を聴かれる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE326
;共通部分
CALL COM_ABLE_COMMON(326)
SIF RESULT == 0
	RETURN 0
;ターゲットが妊娠していないと不可
SIF !TALENT:(MTAR:0):妊娠
	RETURN 0
;プレイヤーが怪我なら不可
SIF CFLAG:(MPLY:0):行動不能状態 == 3
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM326
;実行判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 0, 0)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 0, 0)
	LOCAL:1 = MPLY:0
ENDIF

;●プレイヤー側の処理
;固定で獲得するソース
SOURCE:(MPLY:0):歓楽 = 50
SOURCE:(MPLY:0):接触 = 20

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MPLY:0, 150)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MPLY:0, 50)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MPLY:0, 50, 50)

;●ターゲット側の処理
;固定で獲得するソース
SOURCE:(MTAR:0):歓楽 = 30
SOURCE:(MTAR:0):接触 = 20

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(MTAR:0, 100)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(MTAR:0, 100)

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_N(MTAR:0, 50, 50)

;失敗
IF TFLAG:コマンド成否 == -1
	TIMES SOURCE:(MPLY:0):歓楽, 0.10
	TIMES SOURCE:(MPLY:0):愛情, 0.50
	TIMES SOURCE:(MTAR:0):歓楽, 0.10
	TIMES SOURCE:(MTAR:0):愛情, 0.50
	SOURCE:(MPLY:0):不満 += 500
	SOURCE:(MTAR:0):不満 += 500
	TFLAG:ムード -= 5
;成功
ELSEIF TFLAG:コマンド成否 == 0

;大成功
ELSE
	TIMES SOURCE:(MPLY:0):歓楽, 2.00
	TIMES SOURCE:(MPLY:0):愛情, 2.00
	TIMES SOURCE:(MTAR:0):歓楽, 2.00
	TIMES SOURCE:(MTAR:0):愛情, 2.00
	TFLAG:ムード += 5
ENDIF

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER326(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 10

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 2)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 300
	IF CFLAG:(ARG:0):好感度 < 0
		LOCAL:0 = -30
	ELSEIF CFLAG:(ARG:0):好感度 < 100
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):好奇心
	CALL COM_ORDER_ELEMENT(ARG:0, "好奇心", 8)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE326
IF CFLAG:(MTAR:0):行動不能状態 == 行動不能_臨月
	LOCALS:0 = すっかり大きくなった
ELSEIF TALENT:(MTAR:0):妊娠
	LOCALS:0 = 明らかに膨らんだ
ELSE
	LOCALS:0 =
ENDIF

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%のお腹に耳を当ててみた
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%に言われて%LOCALS:0%お腹に耳を当ててみた
	ENDIF

	;最後のやつ死に分岐になるけどまぁいいや
	IF CFLAG:(MTAR:0):行動不能状態 == 行動不能_臨月
		PRINTFORML 時折お腹の中の子が動く音が聞こえる…
	ELSEIF TALENT:(MTAR:0):妊娠
		PRINTFORML わずかに子供の動く音が聞こえた気がする…
	ELSE
		PRINTFORML さすがにまだ早すぎたのか、特に何も聞こえなかった…
	ENDIF
;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%に%LOCALS:0%お腹の音を聴かせてみた
	;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%のお腹に耳を当てた
	ENDIF

	IF CFLAG:(MTAR:0):行動不能状態 == 行動不能_臨月
		PRINTFORML 時折お腹の中の子が動く音が聞こえるらしく、%ANAME(MPLY:0)%は楽しそうだ…
	ELSEIF TALENT:(MTAR:0):妊娠
		PRINTFORML %ANAME(MPLY:0)%は何か聞こえるのか、真剣な表情で耳を澄ませている…
	ELSE
		PRINTFORML %ANAME(MPLY:0)%はじっと耳を澄ませているが、特に何も聞こえなかったようだ…
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST326
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN326
RETURN コマンド_日常


code:
;スキンシップ

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME327
RESULTS:0 = 抱きしめる
RESULTS:1 = 抱きしめる
RESULTS:2 = ハグをねだる
RESULTS:3 = 抱きしめられる

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE327
;共通部分
CALL COM_ABLE_COMMON(327)
SIF RESULT == 0
	RETURN 0
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM != 1
	RETURN 0
;主人公以外が実行する場合、好感度が3000以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 3000
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0

SIF !CAN_REACH_BODY(MPLY:0, MTAR:0)
	RETURN 0

;プレイヤーが拘束されているならダメ
SIF IS_BIND(MPLY)
	RETURN 0

RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM327
;実行判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

;温泉なら裸でのスキンシップ
;恋人or烙印で濃厚なスキンシップ
IF TALENT:(LOCAL:1):恋人 || TALENT:(LOCAL:1):烙印
	TFLAG:17 = 1

	;レズ・ＢＬ経験基準値
	TFLAG:レズ・ＢＬ経験基準値 = 1
ELSE
	TFLAG:17 = 0
ENDIF

;MPLYについて処理
FOR LOCAL, 0, MPLY_NUM
	;主導権に応じた優越・受動のソース追加
	CALL ADD_SOURCE_INITIATIVE_N(MPLY:LOCAL, 100, 100)
	SOURCE:(MPLY:LOCAL):接触 = 220
	;親密に応じた歓楽のソース追加
	SOURCE:(MPLY:LOCAL):歓楽 = -30
	CALL ADD_SOURCE_KANRAKU(MPLY:LOCAL, 170)
	SOURCE:(MPLY:LOCAL):歓楽 = MAX(SOURCE:(MPLY:LOCAL):歓楽, 0)
	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MPLY:LOCAL, 100)
	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_U(MPLY:LOCAL, 20, 20)
	;成否による修正
	;失敗
	IF TFLAG:コマンド成否 == -1
		TIMES SOURCE:(MPLY:LOCAL):歓楽, 0.10
		TIMES SOURCE:(MPLY:LOCAL):優越, 0.50
		TIMES SOURCE:(MPLY:LOCAL):愛情, 0.50
		SOURCE:(MPLY:LOCAL):不満 += 500
		TCVAR:(MPLY:LOCAL):HOムード値 -= 5
		;成功
	ELSEIF TFLAG:コマンド成否 == 0
		TCVAR:(MPLY:LOCAL):HOムード値 += 2
		;大成功
	ELSE
		TIMES SOURCE:(MPLY:LOCAL):歓楽, 2.00
		TIMES SOURCE:(MPLY:LOCAL):優越, 2.00
		TIMES SOURCE:(MPLY:LOCAL):愛情, 2.00
		TCVAR:(MPLY:LOCAL):HOムード値 += 5
	ENDIF
NEXT
;MTARに付いて処理
FOR LOCAL, 0, MTAR_NUM
	;主導権に応じた優越・受動のソース追加
	CALL ADD_SOURCE_INITIATIVE_N(MTAR:LOCAL, 100, 100)
	SOURCE:(MTAR:LOCAL):接触 = 180
	;親密に応じた歓楽のソース追加
	SOURCE:(MTAR:LOCAL):歓楽 = -30
	CALL ADD_SOURCE_KANRAKU(MTAR:LOCAL, 170)
	SOURCE:(MTAR:LOCAL):歓楽 = MAX(SOURCE:(MTAR:LOCAL):歓楽, 0)
	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MTAR:LOCAL, 100)
	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_U(MTAR:LOCAL, 20, 20)
	;成否による修正
	;失敗
	IF TFLAG:コマンド成否 == -1
		TIMES SOURCE:(MTAR:LOCAL):歓楽, 0.10
		TIMES SOURCE:(MTAR:LOCAL):優越, 0.50
		TIMES SOURCE:(MTAR:LOCAL):愛情, 0.50
		SOURCE:(MTAR:LOCAL):不満 += 500
		TCVAR:(MTAR:LOCAL):HOムード値 -= 5
		;成功
	ELSEIF TFLAG:コマンド成否 == 0
		TCVAR:(MTAR:LOCAL):HOムード値 += 2
		;大成功
	ELSE
		TIMES SOURCE:(MTAR:LOCAL):歓楽, 2.00
		TIMES SOURCE:(MTAR:LOCAL):優越, 2.00
		TIMES SOURCE:(MTAR:LOCAL):愛情, 2.00
		TCVAR:(MTAR:LOCAL):HOムード値 += 5
	ENDIF
NEXT

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP327
RETURN 1

;-------------------------------------------------
;継続状態の処理
;-------------------------------------------------
@COM_EQUIP327(ARG:0)
LOCAL:2 = MEQUIP_PLAYER:(ARG:0):0
LOCAL:3 = MEQUIP_TARGET:(ARG:0):0

;ソースを退避
CALL PUTOUT_SOURCE(LOCAL:2)
CALL PUTOUT_SOURCE(LOCAL:3)

;●プレイヤーの処理
SOURCE:(LOCAL:2):歓楽 = -25 / 4
SOURCE:(LOCAL:2):接触 = 200 / 4
SOURCE:(LOCAL:2):愛情 = -54 / 4
SOURCE:(LOCAL:2):性行動 = 15

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(LOCAL:2, 50 / 4)
SOURCE:(LOCAL:2):歓楽 = MAX(SOURCE:(LOCAL:2):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(LOCAL:2, 180 / 4)
SOURCE:(LOCAL:2):愛情 = MAX(SOURCE:(LOCAL:2):愛情, 0)

IF LOCAL:2 != MASTER && !TALENT:(LOCAL:2):恋人
	SOURCE:(LOCAL:2):逸脱 = 200 / 4
ENDIF

IF LOCAL:2 != MASTER
	IF CFLAG:(LOCAL:2):好感度 < 300
		SOURCE:(LOCAL:2):反感 = 220
	ELSEIF CFLAG:(LOCAL:2):好感度 < 500
		SOURCE:(LOCAL:2):反感 = 120
	ELSEIF CFLAG:(LOCAL:2):好感度 < 800
		SOURCE:(LOCAL:2):反感 = 20
	ENDIF
ENDIF

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_U(LOCAL:2, 20, 20)

;倒錯度変化基準値
TCVAR:(LOCAL:2):倒錯度変化基準値 -= 1

;●ターゲットの処理
SOURCE:(LOCAL:3):歓楽 = -25 / 4
SOURCE:(LOCAL:3):接触 = 200 / 4
SOURCE:(LOCAL:3):愛情 = -54 / 4
SOURCE:(LOCAL:3):性行動 = 15

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(LOCAL:3, 50 / 4)
SOURCE:(LOCAL:3):歓楽 = MAX(SOURCE:(LOCAL:3):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(LOCAL:3, 180 / 4)
SOURCE:(LOCAL:3):愛情 = MAX(SOURCE:(LOCAL:3):愛情, 0)

IF LOCAL:3 != MASTER && !TALENT:(LOCAL:3):恋人
	SOURCE:(LOCAL:3):逸脱 = 200 / 4
ENDIF

IF LOCAL:3 != MASTER
	IF CFLAG:(LOCAL:3):好感度 < 300
		SOURCE:(LOCAL:3):反感 = 220
	ELSEIF CFLAG:(LOCAL:3):好感度 < 500
		SOURCE:(LOCAL:3):反感 = 120
	ELSEIF CFLAG:(LOCAL:3):好感度 < 800
		SOURCE:(LOCAL:3):反感 = 20
	ENDIF
ENDIF

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_U(LOCAL:3, 20, 20)

;倒錯度変化基準値
TCVAR:(LOCAL:3):倒錯度変化基準値 -= 1

;退避したソースを加算
CALL SUM_SOURCE(LOCAL:2)
CALL SUM_SOURCE(LOCAL:3)

;-------------------------------------------------
;継続中の表示
;-------------------------------------------------
@EQUIP_MESSAGE327(ARG:0)
RESULTS = %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%を抱きしめている

;-------------------------------------------------
;継続中の地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE_EQUIP327(ARG:0)
PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%を抱きしめている…

;-------------------------------------------------
;継続を解除したときの地の文
;-------------------------------------------------
@COM_TEXT_RELEASE_EQUIP327(ARG:0)
PRINTFORMW %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%を抱く腕を離した…

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER327(ARG:0)
CALL COM_ORDER_COMMON327(ARG:0)
RETURN 1

@COM_ORDER_TARGET327(ARG:0)
CALL COM_ORDER_COMMON327(ARG:0)
RETURN 1

@COM_ORDER_COMMON327(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 42

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 2)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):貞操観念
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操観念", -4)
ENDIF
IF TALENT:(ARG:0):貞操無頓着
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操無頓着", 3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 5)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 4)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE327
;キス中
IF IS_EQUIP(MPLY:0, 20, 342)
	LOCALS:0 = 唇を重ねながら
	;地の文カット
	LOCAL:0 = SEARCH_EQUIP_IC(342, MPLY:0, MTAR:0)
	IF LOCAL:0 >= 0
		MEQUIP_CUT:(LOCAL:0) = 1
	ENDIF
ELSE
	LOCALS:0 =
ENDIF

IF TFLAG:17 == 1
	LOCALS:1 = 強く
ENDIF

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%を%LOCALS:1%抱きしめた
		;主導権が相手側
	ELSE
		PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の求めに応じて%LOCALS:1%抱きしめた
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の胸に顔をうずめ嬉しそうにしている…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML %ANAME(MTAR:0)%はいささ居心地悪そうに苦笑した…
	ENDIF
	;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の求めに応じて、%LOCALS:0%%LOCALS:1%抱きしめた
		;主導権が相手側
	ELSE
		IF IS_SERVEAT(MPLY:0, "口", "フェラ")
			PRINTFORMW %ANAME(MPLY:0)%は%LOCALS:0%%ANAME(MTAR:0)%の身体に%LOCALS:1%抱きつき、股間に顔を埋めて口淫に耽っている
		ENDIF
	ENDIF

	IF TFLAG:コマンド成否 == 1
		PRINTFORML %ANAME(MPLY:0)%はとても嬉しそうに抱擁する腕に力を込めた…
	ELSEIF TFLAG:コマンド成否 == -1
		PRINTFORML %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の反応が淡白だったせいで拗ねてしまったようだ…
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST327
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN327
IF FLAG:調教モード != 調教_居住区 && FLAG:調教モード != 調教_クエスト
	RETURN コマンド_ウフフ
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;スキンシップ

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME328
RESULTS:0 = 膝枕する
RESULTS:1 = 膝枕する
RESULTS:2 = 膝枕をねだる
RESULTS:3 = 膝枕される

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE328
;共通部分
CALL COM_ABLE_COMMON(328)
SIF RESULT == 0
	RETURN 0
;プレイヤーとターゲットがそれぞれ1人
SIF MPLY_NUM != 1 || MTAR_NUM != 1
	RETURN 0
;主人公以外が実行する場合、好感度が3000以上必要
SIF MPLY:0 != MASTER && CFLAG:(MPLY:0):好感度 < 3000
	RETURN 0
;怪我なら不可
SIF CFLAG:(MTAR:0):行動不能状態 == 3
	RETURN 0

SIF !CAN_REACH_BODY(MPLY:0, MTAR:0)
	RETURN 0
;膝枕されているなら無理
SIF IS_EQUIP_TARGET(MPLY:0, 328)
	RETURN 0
SIF !C_COND("設備属性:椅子,設備属性:ベッド", "OR", MPLY:0)
	RETURN 0

RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM328
;実行判定
CALL COM_ORDER_COMMON
IF RESULT == 0
	RETURN 0
ENDIF

IF MPLY:0 == MASTER
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MTAR:0, 5, 10)
	LOCAL:1 = MTAR:0
ELSE
	;コマンドの成否をTFLAG:コマンド成否にセット
	CALL JUDGE_COM_RESULT(MPLY:0, 5, 10)
	LOCAL:1 = MPLY:0
ENDIF

;温泉なら裸でのスキンシップ
;恋人or烙印で濃厚なスキンシップ
IF TALENT:(LOCAL:1):恋人 || TALENT:(LOCAL:1):烙印
	TFLAG:17 = 1

	;レズ・ＢＬ経験基準値
	TFLAG:レズ・ＢＬ経験基準値 = 1
ELSE
	TFLAG:17 = 0
ENDIF

;MPLYについて処理
FOR LOCAL, 0, MPLY_NUM
	;主導権に応じた優越・受動のソース追加
	CALL ADD_SOURCE_INITIATIVE_N(MPLY:LOCAL, 100, 100)
	SOURCE:(MPLY:LOCAL):接触 = 150
	;親密に応じた歓楽のソース追加
	SOURCE:(MPLY:LOCAL):歓楽 = -30
	CALL ADD_SOURCE_KANRAKU(MPLY:LOCAL, 150)
	SOURCE:(MPLY:LOCAL):歓楽 = MAX(SOURCE:(MPLY:LOCAL):歓楽, 0)
	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MPLY:LOCAL, 125)
	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_U(MPLY:LOCAL, 20, 20)
	;成否による修正
	;失敗
	IF TFLAG:コマンド成否 == -1
		TIMES SOURCE:(MPLY:LOCAL):歓楽, 0.10
		TIMES SOURCE:(MPLY:LOCAL):優越, 0.50
		TIMES SOURCE:(MPLY:LOCAL):愛情, 0.50
		SOURCE:(MPLY:LOCAL):不満 += 500
		TCVAR:(MPLY:LOCAL):HOムード値 -= 5
		;成功
	ELSEIF TFLAG:コマンド成否 == 0
		TCVAR:(MPLY:LOCAL):HOムード値 += 2
		;大成功
	ELSE
		TIMES SOURCE:(MPLY:LOCAL):歓楽, 2.00
		TIMES SOURCE:(MPLY:LOCAL):優越, 2.00
		TIMES SOURCE:(MPLY:LOCAL):愛情, 2.00
		TCVAR:(MPLY:LOCAL):HOムード値 += 5
	ENDIF
NEXT
;MTARに付いて処理
FOR LOCAL, 0, MTAR_NUM
	;主導権に応じた優越・受動のソース追加
	CALL ADD_SOURCE_INITIATIVE_N(MTAR:LOCAL, 100, 100)
	SOURCE:(MTAR:LOCAL):接触 = 150
	;親密に応じた歓楽のソース追加
	SOURCE:(MTAR:LOCAL):歓楽 = -30
	CALL ADD_SOURCE_KANRAKU(MTAR:LOCAL, 150)
	SOURCE:(MTAR:LOCAL):歓楽 = MAX(SOURCE:(MTAR:LOCAL):歓楽, 0)
	;親密に応じた愛情のソース追加
	CALL ADD_SOURCE_AIZYOU(MTAR:LOCAL, 125)
	;主導権に応じたソースの追加
	CALL ADD_SOURCE_INITIATIVE_U(MTAR:LOCAL, 20, 20)
	;成否による修正
	;失敗
	IF TFLAG:コマンド成否 == -1
		TIMES SOURCE:(MTAR:LOCAL):歓楽, 0.10
		TIMES SOURCE:(MTAR:LOCAL):優越, 0.50
		TIMES SOURCE:(MTAR:LOCAL):愛情, 0.50
		SOURCE:(MTAR:LOCAL):不満 += 500
		TCVAR:(MTAR:LOCAL):HOムード値 -= 4
		;成功
	ELSEIF TFLAG:コマンド成否 == 0
		TCVAR:(MTAR:LOCAL):HOムード値 += 2
		;大成功
	ELSE
		TIMES SOURCE:(MTAR:LOCAL):歓楽, 2.00
		TIMES SOURCE:(MTAR:LOCAL):優越, 2.00
		TIMES SOURCE:(MTAR:LOCAL):愛情, 2.00
		TCVAR:(MTAR:LOCAL):HOムード値 += 4
	ENDIF
NEXT

;主導度変化基準値
TFLAG:主導度変化基準値 = 3

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP328
RETURN 1

;-------------------------------------------------
;継続状態の処理
;-------------------------------------------------
@COM_EQUIP328(ARG:0)
LOCAL:2 = MEQUIP_PLAYER:(ARG:0):0
LOCAL:3 = MEQUIP_TARGET:(ARG:0):0

;ソースを退避
CALL PUTOUT_SOURCE(LOCAL:2)
CALL PUTOUT_SOURCE(LOCAL:3)

;●プレイヤーの処理
SOURCE:(LOCAL:2):歓楽 = -25 / 4
SOURCE:(LOCAL:2):接触 = 180 / 4
SOURCE:(LOCAL:2):愛情 = -64 / 4

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(LOCAL:2, 50 / 4)
SOURCE:(LOCAL:2):歓楽 = MAX(SOURCE:(LOCAL:2):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(LOCAL:2, 180 / 4)
SOURCE:(LOCAL:2):愛情 = MAX(SOURCE:(LOCAL:2):愛情, 0)

IF LOCAL:2 != MASTER && !TALENT:(LOCAL:2):恋人
	SOURCE:(LOCAL:2):逸脱 = 200 / 4
ENDIF

IF LOCAL:2 != MASTER
	IF CFLAG:(LOCAL:2):好感度 < 300
		SOURCE:(LOCAL:2):反感 = 220
	ELSEIF CFLAG:(LOCAL:2):好感度 < 500
		SOURCE:(LOCAL:2):反感 = 120
	ELSEIF CFLAG:(LOCAL:2):好感度 < 800
		SOURCE:(LOCAL:2):反感 = 20
	ENDIF
ENDIF

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_U(LOCAL:2, 20, 20)

;倒錯度変化基準値
TCVAR:(LOCAL:2):倒錯度変化基準値 -= 1

;●ターゲットの処理
SOURCE:(LOCAL:3):歓楽 = -25 / 4
SOURCE:(LOCAL:3):接触 = 200 / 4
SOURCE:(LOCAL:3):愛情 = -54 / 4
SOURCE:(LOCAL:3):性行動 = 15

;親密に応じた歓楽のソース追加
CALL ADD_SOURCE_KANRAKU(LOCAL:3, 50 / 4)
SOURCE:(LOCAL:3):歓楽 = MAX(SOURCE:(LOCAL:3):歓楽, 0)

;親密に応じた愛情のソース追加
CALL ADD_SOURCE_AIZYOU(LOCAL:3, 180 / 4)
SOURCE:(LOCAL:3):愛情 = MAX(SOURCE:(LOCAL:3):愛情, 0)

IF LOCAL:3 != MASTER && !TALENT:(LOCAL:3):恋人
	SOURCE:(LOCAL:3):逸脱 = 200 / 4
ENDIF

IF LOCAL:3 != MASTER
	IF CFLAG:(LOCAL:3):好感度 < 300
		SOURCE:(LOCAL:3):反感 = 220
	ELSEIF CFLAG:(LOCAL:3):好感度 < 500
		SOURCE:(LOCAL:3):反感 = 120
	ELSEIF CFLAG:(LOCAL:3):好感度 < 800
		SOURCE:(LOCAL:3):反感 = 20
	ENDIF
ENDIF

;主導権に応じたソースの追加
CALL ADD_SOURCE_INITIATIVE_U(LOCAL:3, 20, 20)

;倒錯度変化基準値
TCVAR:(LOCAL:3):倒錯度変化基準値 -= 1

;退避したソースを加算
CALL SUM_SOURCE(LOCAL:2)
CALL SUM_SOURCE(LOCAL:3)

;-------------------------------------------------
;継続中の表示
;-------------------------------------------------
@EQUIP_MESSAGE328(ARG:0)
RESULTS = %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%を膝枕している

;-------------------------------------------------
;継続中の地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE_EQUIP328(ARG:0)
PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%を膝枕している

;-------------------------------------------------
;継続を解除したときの地の文
;-------------------------------------------------
@COM_TEXT_RELEASE_EQUIP328(ARG:0)

;-------------------------------------------------
;固有の実行判定
;-------------------------------------------------
@COM_ORDER_PLAYER328(ARG:0)
CALL COM_ORDER_COMMON328(ARG:0)
RETURN 1

@COM_ORDER_TARGET328(ARG:0)
CALL COM_ORDER_COMMON328(ARG:0)
RETURN 1

@COM_ORDER_COMMON328(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 38

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 2)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 3)

;好感度
IF CFLAG:(ARG:0):好感度 < 400
	IF CFLAG:(ARG:0):好感度 < 50
		LOCAL:0 = -40
	ELSEIF CFLAG:(ARG:0):好感度 < 150
		LOCAL:0 = -20
	ELSEIF CFLAG:(ARG:0):好感度 < 300
		LOCAL:0 = -10
	ELSE
		LOCAL:0 = -5
	ENDIF
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度不足", LOCAL:0)
ELSE
	LOCAL:0 = CFLAG:(ARG:0):好感度 / 75
	CALL COM_ORDER_ELEMENT(ARG:0, @"好感度", LOCAL:0)
ENDIF

IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -3)
ENDIF
IF TALENT:(ARG:0):貞操観念
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操観念", -4)
ENDIF
IF TALENT:(ARG:0):貞操無頓着
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操無頓着", 3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 5)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 4)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE328
#DIMS DYNAMIC 実行名
#DIMS DYNAMIC 対象名

実行名 = %ANAME(MPLY:0)%
対象名 = %ANAME(MTAR:0)%

;プレイヤーが主人公側
IF MPLY:0 == MASTER
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %実行名%は%対象名%を膝枕した
		;主導権が相手側
	ELSE
		PRINTFORMW %実行名%は%対象名%の求めに応じて膝枕した
	ENDIF
	;プレイヤーが相手側
ELSE
	;主導権が主人公側
	IF IS_INITIATIVE(MASTER)
		PRINTFORMW %実行名%は%対象名%の求めに応じて、膝枕した
		;主導権が相手側
	ELSE
		PRINTFORMW %実行名%は%対象名%を膝枕した
	ENDIF
ENDIF

;-------------------------------------------------
;地の文(パラメータ・刻印変動後)
;-------------------------------------------------
@COM_TEXT_LAST328
;日常コマンドに共通の地の文
CALL COM_TEXT_LAST_LIFE

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------
@COM_AVAILABLE_WHEN328
IF FLAG:調教モード != 調教_居住区
	RETURN コマンド_日常
ELSE
	RETURN コマンド_居住区
ENDIF


code:
;対面座位

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME32
IF IS_INSERT_SINGLE(MTAR:0, "Ａ") && GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 2, 4, 6, 30) && IS_INSERT_MUTUAL(MPLY:0, MTAR:0) != 2
	LOCALS:0 = 二穴対面座位
ELSE
	LOCALS:0 = 対面座位
ENDIF

RESULTS:0 = %LOCALS:0%で挿入
RESULTS:1 = %LOCALS:0%で挿入させられる
RESULTS:2 = %LOCALS:0%で挿入させる
RESULTS:3 = %LOCALS:0%で挿入される
RESULTS:4 = %LOCALS:0%で挿入させる
RESULTS:5 = %LOCALS:0%見せつけ

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE32
;共通部分
CALL COM_ABLE_COMMON(32)
SIF RESULT == 0
	RETURN 0
;プレイヤーは最大で1人まで
SIF MPLY_NUM <= 0 || MPLY_NUM > 1
	RETURN 0
;ターゲットは最大で1人まで
SIF MTAR_NUM <= 0 || MTAR_NUM > 1
	RETURN 0
;対象にＶが必要
SIF !HAS_VAGINA(MTAR:0, 1)
	RETURN 0
SIF IS_V_HOLD(MTAR:0)
	RETURN 0
;being fucked in the ass, unless it's doggy, sitting backwards, or reverse cowgirl
SIF IS_INSERT_SINGLE(MTAR:0, "Ａ") && !GROUPMATCH(GET_SEX_POSITION_SINGLE(MTAR:0, "Ａ"), 2, 4, 6, 30)
	RETURN 0
SIF !CAN_FUCK(MPLY:0, MTAR:0, 32)
	RETURN 0
;receiving cunnilingus, BJ, rimjob, glans blowjob
SIF IS_EQUIP_TARGET(MPLY:0, 2, 11, 8, 107) || IS_EQUIP_TARGET(MTAR:0, 2, 11, 8, 107)
	RETURN 0
;giving facefuck
SIF IS_EQUIP_PLAYER(MPLY:0, 90) || IS_EQUIP_PLAYER(MTAR:0, 90)
	RETURN 0
RETURN 1

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM32
;実行判定
CALL COM_ORDER_COMMON
SIF RESULT == 0
	RETURN 0

;既に挿入中なら現在の体位を取得
TFLAG:17 = 0
IF IS_INSERT_MUTUAL(MPLY:0, MTAR:0) == 1
	SELECTCASE GET_SEX_POSITION(MPLY:0, MTAR:0)
		CASE 1
			TFLAG:17 = 2
		CASE 2
			TFLAG:17 = 3
		CASE 3
			TFLAG:17 = 1
		CASE 4
			TFLAG:17 = 3
		CASE 5
			TFLAG:17 = 2
		CASE 6
			TFLAG:17 = 3
		CASE 7
			TFLAG:17 = 3
		CASE 8
			TFLAG:17 = 2
	ENDSELECT
ENDIF
IF TFLAG:17 == 0
	SELECTCASE PREVCOM
		CASE 30
			TFLAG:17 = 2
		CASE 31
			TFLAG:17 = 3
		CASE 32
			TFLAG:17 = 1
		CASE 33
			TFLAG:17 = 3
		CASE 34
			TFLAG:17 = 2
		CASE 35
			TFLAG:17 = 3
		CASE 37
			TFLAG:17 = 3
		CASE 38
			TFLAG:17 = 2
	ENDSELECT
ENDIF

;挿入関係を一時リセット
CALL CLEAR_INSERT_FLAG(MPLY:0, "Ｐ")

;●プレイヤー側の処理
DOWNBASE:(MPLY:0):体力 += 140

EXP:(MPLY:0):性技経験値 += 1
EXP:(MPLY:0):性交経験値 += 2

SOURCE:(MPLY:0):快Ｐ = SENSE_SEX_PLAYER_P(MTAR:0, MPLY:0, 900)
SOURCE:(MPLY:0):露出 = 50
SOURCE:(MPLY:0):性行動 = 450

IF MTAR:0 == MASTER
	CALL ADD_SOURCE_AIZYOU(MPLY:0, 130)
ELSEIF MPLY:0 != MASTER
	SOURCE:(MPLY:0):露出 += 200
ENDIF

;性交系の共通処理
CALL COM_COMMON_SEX_P(MPLY:0, MTAR:0)

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 150, 80)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MPLY:0):奉仕経験取得可能 = 1

;●ターゲット側の処理
DOWNBASE:(MTAR:0):体力 += 100

EXP:(MTAR:0):性技経験値 += 1
EXP:(MTAR:0):性交経験値 += 2

SOURCE:(MTAR:0):快Ｃ = SENSE_SEX_TARGET(MPLY:0, MTAR:0, 300)
SOURCE:(MTAR:0):快Ｖ = SENSE_SEX_TARGET(MPLY:0, MTAR:0, 900)
SOURCE:(MTAR:0):露出 = 200
SOURCE:(MTAR:0):性行動 = 450

IF MPLY:0 == MASTER
	CALL ADD_SOURCE_AIZYOU(MTAR:0, 130)
ELSEIF MTAR:0 != MASTER
	SOURCE:(MTAR:0):露出 += 200
ENDIF

;性交系の共通処理
CALL COM_COMMON_SEX_V(MTAR:0, MPLY:0)

;快Ｃの補正
IF PALAM:(MTAR:0):潤滑 < PALAMLV:1
	TIMES SOURCE:(MTAR:0):快Ｃ, 0.30
ELSEIF PALAM:(MTAR:0):潤滑 < PALAMLV:3
	TIMES SOURCE:(MTAR:0):快Ｃ, 0.70
ELSEIF PALAM:(MTAR:0):潤滑 < PALAMLV:5
	TIMES SOURCE:(MTAR:0):快Ｃ, 1.00
ELSEIF PALAM:(MTAR:0):潤滑 < PALAMLV:7
	TIMES SOURCE:(MTAR:0):快Ｃ, 1.20
ELSEIF PALAM:(MTAR:0):潤滑 >= PALAMLV:7
	TIMES SOURCE:(MTAR:0):快Ｃ, 1.40
ENDIF

IF MPLY:0 == MASTER
	IF CFLAG:(MTAR:0):好感度 <= 100
		TIMES SOURCE:(MTAR:0):快Ｃ, 0.50
	ELSEIF CFLAG:(MTAR:0):好感度 <= 300
		TIMES SOURCE:(MTAR:0):快Ｃ, 0.70
	ELSEIF CFLAG:(MTAR:0):好感度 <= 500
		TIMES SOURCE:(MTAR:0):快Ｃ, 0.80
	ELSEIF CFLAG:(MTAR:0):好感度 <= 800
		TIMES SOURCE:(MTAR:0):快Ｃ, 1.00
	ELSEIF CFLAG:(MTAR:0):好感度 <= 1500
		TIMES SOURCE:(MTAR:0):快Ｃ, 1.30
	ELSE
		SOURCE:(MTAR:0):快Ｃ = SOURCE:(MTAR:0):快Ｃ * ((CFLAG:(MTAR:0):好感度 - 1500) / 200 + 150) / 100
	ENDIF
ENDIF

;主導権に応じた優越・屈従のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 150, 100)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MTAR:0):奉仕経験取得可能 = 1

;体位フラグのセット
TCVAR:(MPLY:0):31 = 3
TCVAR:(MTAR:0):33 = 3

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = -1

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 7

RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP32
RETURN 1

;-------------------------------------------------
;継続状態の処理
;-------------------------------------------------
@COM_EQUIP32(ARG:0)
LOCAL:2 = MEQUIP_PLAYER:(ARG:0):0
LOCAL:3 = MEQUIP_TARGET:(ARG:0):0

;ソースを退避
CALL PUTOUT_SOURCE(LOCAL:2)
CALL PUTOUT_SOURCE(LOCAL:3)

DOWNBASE:(LOCAL:2):体力 += 20

EXP:(LOCAL:2):性技経験値 += 1
EXP:(LOCAL:2):性交経験値 += 1

SOURCE:(LOCAL:2):快Ｐ += SENSE_SEX_PLAYER_P(LOCAL:3, LOCAL:2, 450)
SOURCE:(LOCAL:2):露出 += 25
SOURCE:(LOCAL:2):性行動 += 150

;性交系の共通処理
CALL COM_COMMON_SEX_P(LOCAL:2, LOCAL:3)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:2):奉仕経験取得可能 = 1

;倒錯度変化基準値
TCVAR:(LOCAL:2):倒錯度変化基準値 -= 1

DOWNBASE:(LOCAL:3):体力 += 20

EXP:(LOCAL:3):性技経験値 += 1
EXP:(LOCAL:3):性交経験値 += 1

SOURCE:(LOCAL:3):快Ｃ += SENSE_SEX_TARGET(LOCAL:2, LOCAL:3, 150)
SOURCE:(LOCAL:3):快Ｖ += SENSE_SEX_TARGET(LOCAL:2, LOCAL:3, 450)
SOURCE:(LOCAL:3):露出 += 100
SOURCE:(LOCAL:3):性行動 += 150

;性交系の共通処理
CALL COM_COMMON_SEX_V(LOCAL:3, LOCAL:2)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(LOCAL:3):奉仕経験取得可能 = 1

;倒錯度変化基準値
TCVAR:(LOCAL:3):倒錯度変化基準値 -= 1

;退避したソースを加算
CALL SUM_SOURCE(LOCAL:2)
CALL SUM_SOURCE(LOCAL:3)

;-------------------------------------------------
;継続中の表示
;-------------------------------------------------
@EQUIP_MESSAGE32(ARG:0)
RESULTS = %EQUIP_PLAYER_ANAME(ARG:0)%が%EQUIP_TARGET_ANAME(ARG:0)%の膣に対面座位で挿入中

;-------------------------------------------------
;継続中の地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE_EQUIP32(ARG:0)
SELECTCASE RAND:2
	CASE 0
		PRINTFORML %EQUIP_TARGET_ANAME(ARG:0)%の膣が%EQUIP_PLAYER_ANAME(ARG:0)%の%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を咥え込んでいる…
	CASE 1
		LOCAL:0 = MEQUIP_TARGET:(ARG:0):0
			;ターゲットが行動不能
		IF !IS_PLAYABLE(LOCAL:0)
			PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は気絶した%EQUIP_TARGET_ANAME(ARG:0)%を抱きしめたまま、激しく腰を突き上げている…
		ELSE
			SELECTCASE GET_COM_INITIATIVE()
					;プレイヤーに主導権
				CASE 0
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%を抱きしめたまま、さらに激しく腰を突き上げている…
					;ターゲットに主導権
				CASE 1
					PRINTFORML %EQUIP_TARGET_ANAME(ARG:0)%は%EQUIP_PLAYER_ANAME(ARG:0)%を抱きしめる腕に力を入れると、自ら快感を貪るように艶めかしく腰を振っている…
					;第三者に主導権
				CASEELSE
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%EQUIP_TARGET_ANAME(ARG:0)%に対面座位で挿入する%EQUIP_PLAYER_ANAME(ARG:0)%に対し、もっと激しく腰を動かすように命令し
					PRINTFORML %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の体を抱きしめ直すと、さらに力を込めて%EQUIP_TARGET_ANAME(ARG:0)%の膣を突き上げている…
			ENDSELECT
		ENDIF
ENDSELECT

;-------------------------------------------------
;継続を解除したときの地の文
;-------------------------------------------------
@COM_TEXT_RELEASE_EQUIP32(ARG:0)
PRINTFORMW %EQUIP_PLAYER_ANAME(ARG:0)%は%EQUIP_TARGET_ANAME(ARG:0)%の膣内から%BAR_NAME(MEQUIP_PLAYER:(ARG:0):0)%を引き抜いた…

;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER32(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 100

;膣性交共通
CALL COM_ORDER_VSEX_P(ARG:0)
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE32
IF IS_PIERCED(MTAR:0, ピアス_ラビア)
	LOCALS:0 = ピアスの取り付けられた
ELSE
	LOCALS:0 =
ENDIF
IF TALENT:(MTAR:0):処女 == 1
	LOCALS:0 = まだ乙女の
ELSEIF PALAM:(MTAR:0):潤滑 >= PALAMLV:7
	LOCALS:0 = 濡れそぼった
ELSE
	LOCALS:0 =
ENDIF
IF IS_OVULATION(MTAR:0)
	LOCALS:0 = %LOCALS:0%危険日の
ENDIF

IF TALENT:(MTAR:0):体格 == 体格_小柄
	LOCALS:0 = %LOCALS:0%幼いワレメ
ELSE
	LOCALS:0 = %LOCALS:0%秘裂
ENDIF

;ペニスバンド装着中
IF IS_EQUIP_PLAYER(MPLY:0, 50)
	;既に挿入済みの場合
	IF TFLAG:17
		SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
			CASE 0
				;前回が対面座位
				IF TFLAG:17 == 1
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を抱きしめたまま、さらに激しく腰を突き上げ%TARGET_STR("た", "てきた")%…
				;正常位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を抱きしめると、再び腰を動かし始めた…
				;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%ANAME(MTAR:0)%を抱きしめて腰を突き上げ%TARGET_STR("た", "てきた")%…
				ENDIF
			;ターゲットに主導権
			CASE 1
				;前回が対面座位
				IF TFLAG:17 == 1
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%を抱きしめる腕に力を入れると、自ら快感を貪るように艶めかしく腰を振っ%PLAYER_STR("た", "てきた")%…
				;正常位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MTAR:0)%は挿入したまま%ANAME(MPLY:0)%の体を抱きしめると、密着したまま自ら快感を貪るように腰を動かし始めた…
				;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の正面に向き直りその体を抱きしめると、密着したまま自ら快感を貪るように腰を動かし始めた…
				ENDIF
			;第三者に主導権
			CASEELSE
				;前回が対面座位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に対面座位で挿入する%ANAME(MPLY:0)%に対し、もっと激しく腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の体を抱きしめ直すと、さらに力を込めて%ANAME(MTAR:0)%の膣を突き上げ始めた…
				;正常位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を対面座位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を抱きしめると、再び腰を動かし始めた…
				;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を対面座位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%ANAME(MTAR:0)%を抱きしめて腰を突き上げた…
				ENDIF
		ENDSELECT

	ELSE
		SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
			CASE 0
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を抱きしめながら張り型を%LOCALS:0%へと挿入し、腰を突き上げ%TARGET_STR("た", "てきた")%…
			;ターゲットに主導権
			CASE 1
				PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%を抱きしめながら張り型を自らの%LOCALS:0%で咥え込み、腰を動かした…
			;第三者に主導権
			CASEELSE
				PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に、対面座位で%ANAME(MTAR:0)%と交わるように命令した
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を抱きしめながら張り型を%LOCALS:0%へと挿入し、腰を突き上げた…
		ENDSELECT
	ENDIF

;ペニスバンド不使用
ELSE
	;既に挿入済みの場合
	IF TFLAG:17
		SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
			CASE 0
				;前回が対面座位
				IF TFLAG:17 == 1
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を抱きしめたまま、さらに激しく腰を突き上げ%TARGET_STR("た", "てきた")%…
				;正常位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を抱きしめると、再び腰を動かし始めた…
				;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%ANAME(MTAR:0)%を抱きしめて腰を突き上げ%TARGET_STR("た", "てきた")%…
				ENDIF
			;ターゲットに主導権
			CASE 1
				;前回が対面座位
				IF TFLAG:17 == 1
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%を抱きしめる腕に力を入れると、艶めかしく腰を振って%ANAME(MPLY:0)%のペニスを擦り上げ%PLAYER_STR("た", "てきた")%…
				;正常位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORMW %ANAME(MTAR:0)%は挿入したまま%ANAME(MPLY:0)%の体を抱きしめると、密着したまま腰を動かして%ANAME(MPLY:0)%のペニスを擦り上げ%PLAYER_STR("た", "てきた")%…
				;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の正面に向き直りその体を抱きしめると、密着したまま腰を動かして%ANAME(MPLY:0)%のペニスを擦り上げ%PLAYER_STR("た", "てきた")%…
				ENDIF
			;第三者に主導権
			CASEELSE
				;前回が対面座位
				IF TFLAG:17 == 1
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MTAR:0)%に対面座位で挿入する%ANAME(MPLY:0)%に対し、もっと激しく腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%の体を抱きしめ直すと、さらに力を込めて%ANAME(MTAR:0)%の膣を突き上げ始めた…
				;正常位・騎乗位
				ELSEIF TFLAG:17 == 2
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を対面座位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を抱きしめると、再び腰を動かし始めた…
				;後背位・背面座位・背面騎乗位
				ELSEIF TFLAG:17 == 3
					PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に対し、体位を対面座位に変えて再び腰を動かすように命令した
					PRINTFORMW %ANAME(MPLY:0)%は挿入したまま%ANAME(MTAR:0)%の体を正面に向けると、%ANAME(MTAR:0)%を抱きしめて腰を突き上げた…
				ENDIF
		ENDSELECT

	ELSE
		SELECTCASE GET_COM_INITIATIVE()
			;プレイヤーに主導権
			CASE 0
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を抱きしめながら%LOCALS:0%へと挿入し、腰を突き上げ%TARGET_STR("た", "てきた")%…
			;ターゲットに主導権
			CASE 1
				PRINTFORMW %ANAME(MTAR:0)%は%ANAME(MPLY:0)%を抱きしめながら腰を降ろして、%LOCALS:0%で%ANAME(MPLY:0)%のペニスを飲み込ん%TARGET_STR("だ", "できた")%…
			;第三者に主導権
			CASEELSE
				PRINTFORML %ANAME(TFLAG:主導権所有者)%は%ANAME(MPLY:0)%に、対面座位で%ANAME(MTAR:0)%と交わるように命令した
				PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%を抱きしめながら%LOCALS:0%へと挿入し、腰を突き上げた…
		ENDSELECT
	ENDIF
ENDIF

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST32
SIF IS_PIERCED(MPLY:0, ピアス_ペニス)
	PRINTFORML %ANAME(MPLY:0)%のペニスに取り付けられたピアスが、%ANAME(MTAR:0)%の膣内をごりごりと抉っていく……

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN32
RETURN コマンド_ウフフ

;--------------------
;性嗜好用関数
;--------------------
@COM_PREFERENCE_PLAYER_32()
RETURN POWER(2, 性的嗜好_性交したい)

@COM_PREFERENCE_TARGET_32()
RETURN POWER(2, 性的嗜好_性交されたい)

@COM_STACK_SPERM_MPLY_TO_MTAR_32()
RETURN 射精部位_膣内


code:
;フェラチオのある日常

;-------------------------------------------------
;コマンド名称
;-------------------------------------------------
@COM_NAME330
SELECTCASE PREVCOM
	CASE 303, 304, 351
		LOCALS:0 = 机下フェラ
	CASEELSE
		LOCALS:0 = フェラ
ENDSELECT

RESULTS:0 = %LOCALS:0%する
RESULTS:1 = %LOCALS:0%させられる
RESULTS:2 = %LOCALS:0%してもらう
RESULTS:3 = %LOCALS:0%される

;-------------------------------------------------
;選択可否判定
;-------------------------------------------------
@COM_ABLE330
;共通部分
CALL COM_ABLE_COMMON(330)
SIF RESULT == 0
	RETURN 0
;ターゲットに竿が必要
SIF !HAS_PENIS(MTAR:0)
	RETURN 0
;行動不能なら不可
SIF !IS_PLAYABLE(MPLY:0)
	RETURN 0
;キス中は不可
SIF IS_EQUIP(MPLY:0, 342)
	RETURN 0
;主人公以外が実行する場合、実行可能でなければ表示されない
IF MPLY:0 != MASTER
	TCVAR:(MPLY:0):判定値 = 0
	SKIPDISP 1
	CALL COM_ORDER_PLAYER330(MPLY:0)
	SKIPDISP 0
	SIF TCVAR:(MPLY:0):判定値 < TCVAR:(MPLY:0):実行値
		RETURN 0
ENDIF
;主導権を握っている側にフェラの知識が必要
LOCAL:0 = 0
IF FLAG:主導権所有者 == -1
	CALL CHECK_COM_KNOWLEDGE(MPLY:0, 11)
	LOCAL:0 += RESULT
	CALL CHECK_COM_KNOWLEDGE(MTAR:0, 11)
	LOCAL:0 += RESULT
ELSE
	CALL CHECK_COM_KNOWLEDGE(FLAG:主導権所有者, 11)
	LOCAL:0 += RESULT
ENDIF
IF !RESULT
	RETURN 0
ENDIF
RETURN 1

;-------------------------------------------------
;快Ｍソースの倍率を取得する関数 ARG:0=PLAYERのキャラ番号
;-------------------------------------------------
@COM330_RATE_M(ARG:0)
#FUNCTION
LOCAL:5 = 1000
SELECTCASE ABL:(ARG:0):奉仕
	CASE 0
		TIMES LOCAL:5, 0.00
		;TIMES SOURCE:(LOCAL:2):不潔, 4.00
	CASE 1
		TIMES LOCAL:5, 0.10
		;TIMES SOURCE:(LOCAL:2):不潔, 2.50
	CASE 2
		TIMES LOCAL:5, 0.30
		;TIMES SOURCE:(LOCAL:2):不潔, 1.50
	CASE 3
		TIMES LOCAL:5, 0.80
		;TIMES SOURCE:(LOCAL:2):不潔, 1.00
	CASE 4
		TIMES LOCAL:5, 1.00
		;TIMES SOURCE:(LOCAL:2):不潔, 0.50
	CASEELSE
		LOCAL:5 = LOCAL:5 * (100 + (ABL:(ARG:0):奉仕 - 5) * 3) / 100
		;TIMES SOURCE:(LOCAL:2):不潔, 0.10
ENDSELECT

SELECTCASE ABL:(ARG:0):性技
	CASE 0
		TIMES LOCAL:5, 1.00
	CASE 1
		TIMES LOCAL:5, 1.10
	CASE 2
		TIMES LOCAL:5, 1.20
	CASE 3
		TIMES LOCAL:5, 1.30
	CASE 4
		TIMES LOCAL:5, 1.40
	CASEELSE
		LOCAL:5 = LOCAL:5 * ((ABL:(ARG:0):性技 - 5) * 2 + 150) / 100
ENDSELECT

RETURNF LOCAL:5

;-------------------------------------------------
;メイン処理
;-------------------------------------------------
@COM330
;実行判定
CALL COM_ORDER_COMMON
SIF RESULT == 0
	RETURN 0

;●プレイヤー側の処理
DOWNBASE:(MPLY:0):体力 += 120

EXP:(MPLY:0):性技経験値 += 1
EXP:(MPLY:0):口淫経験 += 1

SOURCE:(MPLY:0):奉仕 = SERVE_HOUSHI(MPLY:0, 400)
SOURCE:(MPLY:0):接触 = 50
SOURCE:(MPLY:0):快Ｍ = 300 * COM330_RATE_M(MPLY:0) / 1000
;SOURCE:(MPLY:0):快Ｍ = 300 * COM11_RATE_M(MPLY:0) / 1000
SOURCE:(MPLY:0):性行動 = 300

;主導権に応じた優越・受動のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MPLY:0, 150, 120)

;奉仕経験値を得られるコマンドのフラグ
TCVAR:(MPLY:0):奉仕経験取得可能 = 1

IF HAS_PENIS(MTAR:0)
	;ペニスへのキス
	CALL KISS_COMMON(MPLY:0, @"%ANAME(MTAR:0)%のペニス", GET_SITUATION_BY_TRAIN_MODE())
ENDIF

;●ターゲット側の処理
LOCAL:4 = SENSE_HOUSHI_P(MPLY:0, MTAR:0, 1500)
IF TALENT:(MPLY:0):舌使い
	TIMES LOCAL:4, 1.50
ENDIF
SOURCE:(MTAR:0):快Ｐ += LOCAL:4

;●ソースの計算
DOWNBASE:(MTAR:0):体力 += 60
SOURCE:(MTAR:0):接触 = 50
SOURCE:(MTAR:0):性行動 = 180
;主導権に応じた優越・受動のソース追加
CALL ADD_SOURCE_INITIATIVE_U(MTAR:0, 150, 70)

;主導度変化基準値
TFLAG:主導度変化基準値 = 2

;倒錯度変化基準値
TFLAG:倒錯度変化基準値 = 2

;レズ・ＢＬ経験基準値
TFLAG:レズ・ＢＬ経験基準値 = 7

;フェラの知識がなければ、獲得する
CALL SET_COM_KNOWLEDGE(MPLY:0, 11)
CALL SET_COM_KNOWLEDGE(MTAR:0, 11)
RETURN 1

;-------------------------------------------------
;継続コマンドかどうかを設定
;-------------------------------------------------
@COM_IS_EQUIP330
RETURN 0

;-------------------------------------------------
;固有の実行判定(プレイヤー側)
;-------------------------------------------------
@COM_ORDER_PLAYER330(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 90

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 4)
CALL COM_ORDER_ELEMENT(ARG:0, @"精愛Lv{ABL:(ARG:0):精愛}", ABL:(ARG:0):精愛 * 3)

LOCAL:0 = GET_PALAMLV(PALAM:(ARG:0):欲情)
CALL COM_ORDER_ELEMENT(ARG:0, @"欲情Lv{LOCAL:0}", MIN(LOCAL:0 * 1, 10))

;ムード
IF TFLAG:ムード < 10
	LOCAL:0 = -10
ELSEIF TFLAG:ムード < 30
	LOCAL:0 = -5
ELSEIF TFLAG:ムード < 50
	LOCAL:0 = 0
ELSEIF TFLAG:ムード < 100
	LOCAL:0 = (TFLAG:ムード - 50) / 10
ELSE
	LOCAL:0 = MIN(10, (TFLAG:ムード - 100) / 50 + 5)
ENDIF

IF TALENT:(ARG:0):キス未経験
	CALL COM_ORDER_ELEMENT(ARG:0, "キス未経験", -7)
ENDIF
IF TALENT:(ARG:0):貞操観念
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操観念", -6)
ENDIF
IF TALENT:(ARG:0):貞操無頓着
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操無頓着", 3)
ENDIF
IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -1)
ENDIF
IF TALENT:(ARG:0):汚臭鈍感
	CALL COM_ORDER_ELEMENT(ARG:0, "汚臭鈍感", 1)
ENDIF
IF TALENT:(ARG:0):汚臭敏感
	CALL COM_ORDER_ELEMENT(ARG:0, "汚臭敏感", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):快感の否定
	CALL COM_ORDER_ELEMENT(ARG:0, "快感の否定", -1)
ENDIF
IF TALENT:(ARG:0):男嫌い
	CALL COM_ORDER_ELEMENT(ARG:0, "男嫌い", -12)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 6)
ENDIF

;合意
IF !TALENT:(ARG:0):合意 && !TALENT:(ARG:0):親友
	CALL COM_ORDER_ELEMENT(ARG:0, "合意なし", -10)
ENDIF
RETURN 1

;-------------------------------------------------
;固有の実行判定(ターゲット側)
;-------------------------------------------------
@COM_ORDER_TARGET330(ARG:0)
;実行値の設定
TCVAR:(ARG:0):実行値 = 90

;共通部分
CALL COM_ORDER(ARG:0)

CALL COM_ORDER_ELEMENT(ARG:0, @"欲望Lv{ABL:(ARG:0):欲望}", ABL:(ARG:0):欲望 * 1)
CALL COM_ORDER_ELEMENT(ARG:0, @"奉仕Lv{ABL:(ARG:0):奉仕}", ABL:(ARG:0):奉仕 * 4)
CALL COM_ORDER_ELEMENT(ARG:0, @"精愛Lv{ABL:(ARG:0):精愛}", ABL:(ARG:0):精愛 * 3)

LOCAL:0 = GET_PALAMLV(PALAM:(ARG:0):欲情)
CALL COM_ORDER_ELEMENT(ARG:0, @"欲情Lv{LOCAL:0}", MIN(LOCAL:0 * 1, 10))

;ムード
IF TFLAG:ムード < 10
	LOCAL:0 = -10
ELSEIF TFLAG:ムード < 30
	LOCAL:0 = -5
ELSEIF TFLAG:ムード < 50
	LOCAL:0 = 0
ELSEIF TFLAG:ムード < 100
	LOCAL:0 = (TFLAG:ムード - 50) / 10
ELSE
	LOCAL:0 = MIN(10, (TFLAG:ムード - 100) / 50 + 5)
ENDIF

IF TALENT:(ARG:0):キス未経験
	CALL COM_ORDER_ELEMENT(ARG:0, "キス未経験", -7)
ENDIF
IF TALENT:(ARG:0):貞操観念
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操観念", -6)
ENDIF
IF TALENT:(ARG:0):貞操無頓着
	CALL COM_ORDER_ELEMENT(ARG:0, "貞操無頓着", 3)
ENDIF
IF TALENT:(ARG:0):恥じらい
	CALL COM_ORDER_ELEMENT(ARG:0, "恥じらい", -1)
ENDIF
IF TALENT:(ARG:0):汚臭鈍感
	CALL COM_ORDER_ELEMENT(ARG:0, "汚臭鈍感", 1)
ENDIF
IF TALENT:(ARG:0):汚臭敏感
	CALL COM_ORDER_ELEMENT(ARG:0, "汚臭敏感", -3)
ENDIF
IF TALENT:(ARG:0):献身的
	CALL COM_ORDER_ELEMENT(ARG:0, "献身的", 6)
ENDIF
IF TALENT:(ARG:0):快感の否定
	CALL COM_ORDER_ELEMENT(ARG:0, "快感の否定", -1)
ENDIF
IF TALENT:(ARG:0):男嫌い
	CALL COM_ORDER_ELEMENT(ARG:0, "男嫌い", -12)
ENDIF
IF TFLAG:デート場所
	CALL COM_ORDER_ELEMENT(ARG:0, "デート中", 6)
ENDIF

;合意
IF !TALENT:(ARG:0):合意 && !TALENT:(ARG:0):親友
	CALL COM_ORDER_ELEMENT(ARG:0, "合意なし", -10)
ENDIF
RETURN 1

;-------------------------------------------------
;地の文(前文)
;-------------------------------------------------
@COM_TEXT_BEFORE330
#DIMS PENIS_NAME
;SKIPDISPしてから呼び出してRESULTを取ったら、キャラ口上からどのパターンか読める…ハズ
LOCAL:3 = 0

SELECTCASE RAND:10
	CASE 0 TO 3
		PENIS_NAME = ペニス
	CASE 4 TO 5
		PENIS_NAME = 怒張
	CASE 6 TO 7
		PENIS_NAME = 肉棒
	CASE 8 TO 9
		PENIS_NAME = 男根
;普通は見れない
	CASEELSE
		PENIS_NAME = 勃起チンポ
ENDSELECT

PRINTFORM %ANAME(MPLY:0)%は

SELECTCASE GET_COM_INITIATIVE()
	;主導権がターゲット側
	CASE 1
		SELECTCASE PREVCOM
			;机下フェラ
			CASE 303,304,351
				LOCAL:3 = 1
				PRINTDATA
					DATAFORM %ANAME(MTAR:0)%に命じられると
					DATAFORM %ANAME(MTAR:0)%の目配せだけで意図を察し、
					DATAFORM %ANAME(MTAR:0)%の手招きに従い、
					DATAFORM %ANAME(MTAR:0)%が言ったとおりに
					DATAFORM %ANAME(MTAR:0)%に仕込まれたとおりに
				ENDDATA
				PRINTDATAL
					DATAFORM 机の下に潜り込み、眼前に突きつけられた%PENIS_NAME%を
					DATAFORM 顔にかかったテーブルクロスの端からこちらの様子をうかがいながら%PENIS_NAME%を
					DATAFORM 口を水ですすぐと音を立てずに%ANAME(MTAR:0)%の足元まで這いより%PENIS_NAME%を
					DATAFORM 途中でぶつけた額をさすりながら、机の下で%PENIS_NAME%を
				ENDDATA
			;蒸れフェラ
			CASE 301, 302
				LOCAL:3 = 2
				PRINTDATA
					DATAFORM 汗でテカリ、
					DATAFORM 訓練の興奮で滾り、
					DATAFORM 激しい運動で蒸れ、
					DATAFORM 訓練の疲労感に勃起し、
				ENDDATA
				PRINTDATAL
					DATAFORM 蒸気の上がっている%ANAME(MTAR:0)%の%PENIS_NAME%を
					DATAFORM 雄の臭いを撒き散らす%ANAME(MTAR:0)%の%PENIS_NAME%を
					DATAFORM 男のフェロモンを立ち上らせる%ANAME(MTAR:0)%の%PENIS_NAME%を
					DATAFORM 激しく反り返っている%ANAME(MTAR:0)%の%PENIS_NAME%を
				ENDDATA
				PRINTFORML
			;歌唱フェラ
			CASE 306
				LOCAL:3 = 3
				PRINTFORML 歌声を奏でる唇に%ANAME(MTAR:0)%の%PENIS_NAME%を突きつけられると、そのまま
			;お掃除フェラ
			CASE 310,311
				LOCAL:3 = 4
				PRINTDATAL
					DATAFORM %ANAME(MTAR:0)%が杯から垂らした雫に導かれるように%PENIS_NAME%に舌を伸ばすと
					DATAFORM 酒のおかわりをやると言って己の%PENIS_NAME%に酒をぶちまけた%ANAME(MTAR:0)%の股座に顔を埋めると
				ENDDATA
			;実地演習
			CASE 550
				LOCAL:3 = 5
				PRINTFORML 早速実践するようにと突きつけられた%ANAME(MTAR:0)%の%PENIS_NAME%を
			;路地裏フェラ
			CASE 305
				LOCAL:3 = 6
				PRINTDATAL
					DATAFORM %ANAME(MTAR:0)%に人気のない路地裏へと連れ込まれ、頬を露出した%PENIS_NAME%で叩かれると
					DATAFORM %ANAME(MTAR:0)%に人気のない路地裏で求められ、せめて口でと、%PENIS_NAME%を頬に押しつけられる
				ENDDATA
			CASE 352
				LOCAL:3 = 8
				PRINTFORML %ANAME(MTAR:0)%に本棚の陰に連れ込まれ、頬を露出した%PENIS_NAME%で叩かれると
			;猥談
			CASE 309
				LOCAL:3 = 9
				PRINTDATAL
					DATAFORM 猥談に興奮した%ANAME(MTAR:0)%に今の話を実演するように求められ、%PENIS_NAME%を眼前に突きつけられると
					DATAFORM 猥談に興奮した%ANAME(MTAR:0)%に押し倒され、%PENIS_NAME%を眼前に突き出されると
					DATAFORM 猥談に興奮した%ANAME(MTAR:0)%に、ちょっと鎮めてくれよと頼まれ、%PENIS_NAME%を露出させると
				ENDDATA
			CASEELSE
				IF GET_PLACENAME(TFLAG:デート場所) == "市街地"
					LOCAL:3 = 6
					PRINTDATAL
						DATAFORM %ANAME(MTAR:0)%に人気のない路地裏へと連れ込まれ、頬を露出した%PENIS_NAME%で叩かれると
						DATAFORM %ANAME(MTAR:0)%に人気のない路地裏で求められ、せめて口でと、%PENIS_NAME%を頬に押しつけられる
					ENDDATA
			;温泉
				ELSEIF GET_PLACENAME(TFLAG:デート場所) == "温泉"
					LOCAL:3 = 10
					PRINTDATAL
						DATAFORM 命じられるがまま、%ANAME(MTAR:0)%の腰を支えると水面から屹立した%PENIS_NAME%を
						DATAFORM %ANAME(MTAR:0)%に命じられ、風呂椅子に座る%ANAME(MTAR:0)%の%PENIS_NAME%を
						DATAFORM 手近な岩に腰掛けた%ANAME(MTAR:0)%に頼まれ、お湯の滴る%PENIS_NAME%を
					ENDDATA
				ELSEIF GET_PLACENAME(TFLAG:デート場所) == "遊泳場"
					LOCAL:3 = 11
					PRINTDATAL
						DATAFORM 泳ぎ疲れ岩に腰掛ける%ANAME(MTAR:0)%に頼まれ、水の滴る%PENIS_NAME%を
						DATAFORM %ANAME(MTAR:0)%に岩陰へ連れ込まれ、%PENIS_NAME%を頬へ押し当てられる
						DATAFORM %ANAME(MTAR:0)%に頼まれ、水中で剥き出しにされた%PENIS_NAME%を
						DATAFORM 泳いで冷えたから暖めてくれと%ANAME(MTAR:0)%に頼まれ、%PENIS_NAME%を
					ENDDATA
				ELSEIF GET_PLACENAME(TFLAG:デート場所) == "森林"
					LOCAL:3 = 10
					PRINTDATAL
						DATAFORM 獲物を追うのに飽きた%ANAME(MTAR:0)%に命じられ、跪いて%PENIS_NAME%を露出させると
						DATAFORM %ANAME(MTAR:0)%に木陰に連れ込まれ、眼前に%PENIS_NAME%を突きつけられると
						DATAFORM %ANAME(MTAR:0)%に森小屋へ連れ込まれ、%PENIS_NAME%への奉仕を命じられると
					ENDDATA
				ELSE
					PRINTDATAL
						DATAFORM 唐突に%PENIS_NAME%を露出させる%ANAME(MTAR:0)%の行動に戸惑いつつも、周囲の気配を伺いながら
						DATAFORM %ANAME(MTAR:0)%に、催したから処理してくれと%PENIS_NAME%を突きつけられる
						DATAFORM %ANAME(MTAR:0)%に物陰に連れ込まれ、眼前に%PENIS_NAME%を突きつけられる
						DATAFORM %ANAME(MTAR:0)%に跪かされ、溜まってきたから処理してくれと%PENIS_NAME%を突き出される
					ENDDATA
				ENDIF
		ENDSELECT
		IF ABL:(MPLY:0):精愛 > 4
			PRINTDATA
				DATAFORM お預けを解かれた犬のように狂喜して、
				DATAFORM 溢れ出る唾と一緒に飲み込むような勢いで、
				DATAFORM 誰にも渡さないとばかりに、
				DATAFORM 口の端からよだれが溢れるのも気にせず、
				DATAFORM 蕩かせた表情を隠すことなく、
				DATAFORM 奉仕できる喜びからか目尻を下げ、
				DATAFORM 頬を紅潮させながら、
				DATAFORM 鼻を通りゆく性臭でスイッチが入ったのか、
				DATAFORM 快楽に崩れた表情であることに気づかぬぐらいに、
				DATAFORM 徐々に快楽に蕩けた表情を見せながら、
				DATAFORM 下品な音を立てていることも気にせず、
			ENDDATA
		ENDIF
		IF TALENT:(MPLY:0):体格 == 体格_小柄
			PRINTDATA
				DATAFORM %PENIS_NAME%が収まらない口で必死ながら
				DATAFORM 薄く可愛らしい唇で
				DATAFORM 小さな口で必死ながら
				DATAFORM
			ENDDATA
		ELSE
			PRINTDATA
				DATAFORM 口一杯に
				DATAFORM 艶めく唇で
				DATAFORM 顔についた肉穴で
				DATAFORM 濡れた唇で
			ENDDATA
		ENDIF
		PRINTDATA
			DATAFORM 咥え込み、
			DATAFORM 頬張ると、
			DATAFORM むしゃぶりつき、
		ENDDATA
		IF ABL:(MPLY:0):性技 >= 3
			PRINTDATA
				DATAFORM 巧みな舌使いで
				DATAFORM 別の穴に入れたかと思うほどの蠕動で
				DATAFORM 巧みな口使いで
				DATAFORM 丁寧かつきめ細やかな口技で
			ENDDATA
			SIF GETBIT(TALENT:(MPLY:0):淫乱系, 素質_淫乱_精液便女)
				PRINTFORM 男を悦ばせるための便女らしく
			PRINTDATAW
				DATAFORM %ANAME(MTAR:0)%を悦ばせた…
				DATAFORM %ANAME(MTAR:0)%をもてなした…
				DATAFORM %ANAME(MTAR:0)%に奉仕した…
				DATAFORM %ANAME(MTAR:0)%の%PENIS_NAME%をねっとりしゃぶり続けた…
			ENDDATA
		ELSE
			PRINTDATA
				DATAFORM 不慣れな舌使いで
				DATAFORM ぎこちない動きで
				DATAFORM 慣れない口使いで
				DATAFORM 時々えづきながら
			ENDDATA
			SIF GETBIT(TALENT:(MPLY:0):淫乱系, 素質_淫乱_精液便女)
				PRINTFORM 男を悦ばせるための便女らしく
			PRINTDATAW
				DATAFORM %ANAME(MTAR:0)%の%PENIS_NAME%を責めた…
				DATAFORM %ANAME(MTAR:0)%への奉仕を続けた…
				DATAFORM %ANAME(MTAR:0)%の%PENIS_NAME%をしゃぶり続けた…
			ENDDATA
		ENDIF
	;主導権がプレイヤー側
	CASE 0
		SELECTCASE PREVCOM
			;机下フェラ
			CASE 303,304,351
				LOCAL:3 = 21
				PRINTFORML こっそりと机の下に潜り込み、%ANAME(MTAR:0)%の%PENIS_NAME%を露出させると
			;蒸れフェラ
			CASE 301,302
				LOCAL:3 = 22
				PRINTFORML 汗で蒸れている%ANAME(MTAR:0)%の下着を剥ぎ取り、蒸気の上がっている%PENIS_NAME%を
			;歌唱フェラ
			CASE 306
				LOCAL:3 = 23
				PRINTFORML 歌を続けながら顔を%ANAME(MTAR:0)%の股間に近づけ、取り出した%PENIS_NAME%に歌いかけるように
			;お掃除フェラ
			CASE 310,311
				LOCAL:3 = 24
				PRINTFORML わざとらしく%ANAME(MTAR:0)%の股間に\@ PREVCOM == 310 ? 酒杯 # エインズの露 \@の残りをこぼすと、濡れた下着ごと%ANAME(MTAR:0)%の%PENIS_NAME%を
			;実地演習
			CASE 550
				LOCAL:3 = 25
				PRINTFORML 早速実践だとばかりに%ANAME(MTAR:0)%の%PENIS_NAME%を取り出すと
			;路地裏フェラ
			CASE 305
				LOCAL:3 = 26
				PRINTFORML 人気のない路地裏へと連れ込むと、%ANAME(MTAR:0)%の下着を剥ぎ取りそのまま%PENIS_NAME%を
			CASE 352
				LOCAL:3 = 28
				PRINTFORML %ANAME(MTAR:0)%を本棚の陰に連れ込み、%ANAME(MTAR:0)%の下着を剥ぎ取りそのまま%PENIS_NAME%を
			;猥談
			CASE 309
				LOCAL:3 = 29
				PRINTFORML 猥談に興奮を抑えられなくなり、%ANAME(MTAR:0)%の下着を剥ぎ取りそのまま%PENIS_NAME%を
			CASEELSE
				IF GET_PLACENAME(TFLAG:デート場所) == "市街地"
					LOCAL:3 = 26
					PRINTFORML 人気のない路地裏へと連れ込むと、%ANAME(MTAR:0)%の下着を剥ぎ取りそのまま%PENIS_NAME%を
			;温泉
				ELSEIF GET_PLACENAME(TFLAG:デート場所) == "温泉"
					LOCAL:3 = 27
					IF RAND:2 == 0
						PRINTFORML %ANAME(MTAR:0)%の腰を浮かせて%PENIS_NAME%を水面から浮上させると
					ELSE
						PRINTFORML %ANAME(MTAR:0)%を手近な岩に腰掛けさせると、お湯が滴る%PENIS_NAME%を
					ENDIF
				ELSEIF GET_PLACENAME(TFLAG:デート場所) == "遊泳場"
					LOCAL:3 = 30
					IF RAND:3 == 0
						PRINTFORML %ANAME(MTAR:0)%の腰を浮かせて%PENIS_NAME%を水面から浮上させると
					ELSEIF RAND:2 == 0
						PRINTFORML %ANAME(MTAR:0)%に絡みつき、水中で%PENIS_NAME%を露出させると
					ELSE
						PRINTFORML %ANAME(MTAR:0)%を手近な岩に腰掛けさせると、水が滴る%PENIS_NAME%を
					ENDIF
				ELSEIF GET_PLACENAME(TFLAG:デート場所) == "森林"
					IF RAND:3 == 0
						PRINTFORML 大きな獲物があったと呟いて、%ANAME(MTAR:0)%の下着を剥ぎ取ると
					ELSE
						PRINTFORML %ANAME(MTAR:0)%を木陰に連れ込むと、%PENIS_NAME%を露出させ
					ENDIF
				ELSE
					LOCAL:3 = 20
					PRINTFORML 周囲に人気がないのを確認すると、%ANAME(MTAR:0)%の%PENIS_NAME%を露出させ
				ENDIF
		ENDSELECT
		IF ABL:(MPLY:0):精愛 > 4
			PRINTDATA
				DATAFORM 我慢の限界といった表情で、
				DATAFORM 半狂乱で頬擦りしてから一気に、
				DATAFORM 愛おしむようにキスの雨を振らせてから、
				DATAFORM むせかえるような雄の匂いに瞳を蕩かして、
				DATAFORM たっぷりと唾液をまぶしながら
			ENDDATA
		ENDIF
		IF TALENT:(MPLY:0):体格 == 体格_小柄
			IF RAND:2 == 0
				PRINTFORM 小さな口で必死に
			ELSE
				PRINTFORM 薄く可愛らしい唇で
			ENDIF
		ELSE
			IF RAND:2 == 0
				PRINTFORM 口一杯に
			ELSE
				PRINTFORM 艶めく唇で
			ENDIF
		ENDIF
		IF RAND:3 == 0
			PRINTFORM 咥え込み
		ELSEIF RAND:2 == 0
			PRINTFORM 頬張り
		ELSE
			PRINTFORM むしゃぶりつき
		ENDIF
		IF GETBIT(TALENT:(MPLY:0):淫乱系, 素質_淫乱_精液便女)
			PRINTFORM 、男を悦ばせるための便女らしく
		ENDIF
		IF RAND:2 == 0
			PRINTFORMW 奉仕している…
		ELSE
			PRINTFORMW 一心不乱に味わっている…
		ENDIF
ENDSELECT

PRINTL

RETURN LOCAL:3

;--------------------------------------------------------
;地の文(パラメータ・刻印変動後)
;--------------------------------------------------------
@COM_TEXT_LAST330

LOCAL:0 = 0
LOCAL:1 = 0

;SKIPDISPしてから呼び出してRESULTを取ったら、キャラ口上からどのパターンか読める…ハズ

LOCAL:3 = 0

;射精させれたかどうか
LOCAL:1 = NOWEX:(MTAR:0):射精

;精愛で分岐
IF ABL:(MPLY:0):精愛 > 4
	LOCAL:0 = 2
ELSEIF ABL:(MPLY:0):精愛 > 1
	LOCAL:0 = 1
ENDIF

SELECTCASE PREVCOM
	;食ザー
	CASE 351
		IF LOCAL:1 > 0 && LOCAL:0 > 1
			LOCAL:3 = 10
			PRINTFORML 何事もないかのように席に戻った%ANAME(MPLY:0)%は、口内に残した%ANAME(MTAR:0)%の精液を料理に垂らすと
			PRINTFORMW とても美味しそうに、一口一口味わいながら完食した
		ELSEIF LOCAL:0 > 1
			LOCAL:3 = 11
			PRINTFORML 名残惜しそうに%ANAME(MTAR:0)%のペニスから口を離して席に戻った%ANAME(MPLY:0)%は
			PRINTFORMW もう一味足したかった、と残りの料理を弄びながら食べ終えた
		ELSEIF LOCAL:0 > 0
			LOCAL:3 = 12
			PRINTFORML %ANAME(MPLY:0)%は席に戻ると、%ANAME(MTAR:0)%の味しかしない、と
			PRINTFORMW 拗ねたように、しかし嬉しそうに愚痴りながら残りの食事を味わった
		ELSE
		ENDIF
	;飲ザー
	CASE 310,311
		IF LOCAL:1 > 0 && LOCAL:0 > 1
			LOCAL:3 = 20
			IF RAND:2 == 0
				PRINTFORML %ANAME(MPLY:0)%はなおも鈴口から迸る%ANAME(MTAR:0)%の精液を杯で受け止め、口をつけた
				PRINTFORMW 味わいつつ最後の一口を飲み込むと、小さな、そして艶やかな溜息をこぼした
			ELSE
				PRINTFORML %ANAME(MPLY:0)%は口内に残していた%ANAME(MTAR:0)%の精液を
				PRINTFORMW 新たな\@ PREVCOM == 310 ? 酒杯 # エインズの露 \@に垂らし、時折かき混ぜながら味の変化を楽しんだ
			ENDIF
		ELSEIF LOCAL:0 > 1
			LOCAL:3 = 21
			PRINTFORML ふやけるほどに%ANAME(MTAR:0)%のペニスを味わい尽くした%ANAME(MPLY:0)%は
			PRINTFORMW 鈴口の前に杯を掲げて、精液で満たして欲しそうに上目遣いで懇願してきた
		ELSEIF LOCAL:0 > 0
			LOCAL:3 = 22
			PRINTFORMW %ANAME(MPLY:0)%は\@ PREVCOM == 310 ? 酒杯 # エインズの露 \@の残りをゆっくりと、一口ずつ味わっては顔を惚けさせながら飲み込んでいった
		ELSEIF (ABL:(MPLY:0):主導度Ｕ <= -100 || CFLAG:(MPLY:0):従属度 >= 1500) && LOCAL:1 > 1
			LOCAL:3 = 23
			PRINTFORML %ANAME(MTAR:0)%は%ANAME(MPLY:0)%の杯になみなみと射精し、飲み干すように促すと
			PRINTFORMW %ANAME(MPLY:0)%は体を震わせながら最後の一滴まで飲み干した
		ELSE
		ENDIF
	;なでなで
	CASE 321
		LOCAL:3 = 30
		PRINTFORMW ペニスを咥えたまま%ANAME(MTAR:0)%に頭を撫でられると、%ANAME(MPLY:0)%は嬉しそうに目を細めた
	;髪で拭う
	CASE 322,323
		IF ABL:(MPLY:0):主導度Ｕ <= -100 || CFLAG:(MPLY:0):従属度 >= 1500
			LOCAL:3 = 40
			PRINTFORML %ANAME(MTAR:0)%は、%ANAME(MPLY:0)%の髪がベタベタになるのも構わずペニスの汚れを拭い取った
			PRINTFORMW 卑猥な光沢を放つ己の髪を%ANAME(MPLY:0)%は頬を染めながらいじっている
		ENDIF
	CASEELSE
	;パパのミルク
	;おっさんくさかったのでコメントアウト
;		IF TALENT:(MPLY:0):妊娠 || (TALENT:(MPLY:0):母乳体質 && CFLAG:(MPLY:0):行動不能状態)
;			IF LOCAL:1 > 0 && LOCAL:0 > 1
;				IF RAND:3 == 0 && TALENT:(MPLY:0):妊娠
;				LOCAL:3 = 50
;					PRINTFORMW お腹の子もパパのミルクが大好きになりますように、と祈る%ANAME(MPLY:0)%は慈母そのものであった…祈りの内容以外は
;				ELSE
;				LOCAL:3 = 51
;					PRINTFORMW 母乳の出が良くなる、と%ANAME(MPLY:0)%は嬉しそうに\@ TALENT:(MPLY:0):妊娠 ? お腹 # 胸 \@をさすっている
;				ENDIF
;			ELSEIF LOCAL:0 > 1
;				LOCAL:3 = 52
;				PRINTFORMW \@ TALENT:(MPLY:0):妊娠 ? お腹の子に # 子供のためにも \@パパのミルクを頂戴？と%ANAME(MPLY:0)%は妖しく微笑んだ
;			ELSEIF LOCAL:0 > 0 && TALENT:(MPLY:0):妊娠
;				LOCAL:3 = 53
;				PRINTFORMW %ANAME(MPLY:0)%がお腹をさすりながら、この子も精液が好きになったらどうしよう…と呟いた
;			ELSE
;				LOCAL:3 = 54
;				PRINTFORMW %ANAME(MPLY:0)%は子供の栄養のため、と%ANAME(MTAR:0)%に諭され困惑の表情を浮かべた
;			ENDIF
	;それ以外
;		ELSE
		IF LOCAL:1 > 0 && LOCAL:0 > 1
			LOCAL:3 = 1
			PRINTFORMW %ANAME(MPLY:0)%は口をもごもごとさせて喉を鳴らした後、%ANAME(MTAR:0)%の耳元でご馳走様と囁いた
		ELSEIF LOCAL:0 > 1
			LOCAL:3 = 2
			PRINTFORMW 名残惜しそうに%ANAME(MTAR:0)%のペニスから口を離した%ANAME(MPLY:0)%は、指を口に当ててナニかを催促してきた
		ELSEIF LOCAL:0 > 0
			LOCAL:3 = 3
			PRINTFORML %ANAME(MPLY:0)%は、%ANAME(MTAR:0)%のペニスをふきベトベトになった手巾で口を拭った
			PRINTFORMW その目は潤み頬は赤みを帯び、口元の緩みはしばし消えることはなかった
		ELSE
			PRINTFORMW %ANAME(MPLY:0)%は%ANAME(MTAR:0)%のペニスと、自分の口を手巾で綺麗に拭い取り身だしなみを整えた
		ENDIF
;		ENDIF
ENDSELECT

RETURN LOCAL:3

;-------------------------------------------------

;コマンド区分
;-------------------------------------------------

@COM_AVAILABLE_WHEN330
RETURN コマンド_日常

@COM_STACK_SPERM_MTAR_TO_MPLY_330()
RETURN RAND:3 ? 射精部位_口 # 射精部位_顔